Most books on Windows start with its guts: how it is a message-driven environment, how each application usually has a main window and a message dispatching loop, and so on. That’s all true, but when using OWL, it doesn’t matter. The message loop is more or less buried in the tApplication class from which your application descends. OWL hides so much of Windows’ guts, in fact, that you might almost be lulled into thinking you don’t have to know anything about them at all. Nothing could be further from the truth. The fact is, programming for Windows is neither easier nor harder than programming for DOS; it’s just different.

To begin, we’re going to start with a simple but useful string class; for as you’ll see, BPW introduces a new wrinkle to the string problem we thought Turbo Pascal for DOS had solved years ago.

If you didn’t come to BPW from Turbo Pascal for Windows, the first jolt you are apt to get comes when you stumble on the pChar data type. “What the heck’s that?” you might well be heard to say.

A String for All Seasons�tc "A String for All Seasons"�

Strings were not part of the original Pascal language as envisioned by its creator, Niklaus Wirth. But programmers needed them, nevertheless, and usually placed them in a “packed array of char.” Microsoft and Borland added a string data type to the language definition, though each with a slightly different syntax. Now that Microsoft Pascal has gone the way of the Edsel, Borland’s syntax reigns and the “packed” keyword no longer has any meaning.

The Turbo (and Borland) Pascal string is a sensible compromise for implementing a string. On the one hand, it would be nice to allow strings of any length. But string operations are most efficient if, somewhere, the length of the string has been kept in a safe place. In PL/I (my favorite mainframe language), each string had an elaborate descriptor associated with it. Included in the descriptor was a maximum length for the string, its current length, and a pointer to the string itself. You could do anything with a PL/I string. It might take a little time, but hey—it’s a mainframe. No one expected efficient execution.

In C, with its neurotic need to provide the programmer with infinite power in as few bytes as possible, the descriptor was thrown out and the array of characters was accessed directly. This meant that most string operations—even one as simple as obtaining the number of characters in a string—could only be accomplished by counting each character until a special end-of-string character was found. This counting slows C string operations considerably. What’s more, C strings cannot contain embedded characters with all zero bits (the NULL character), as it’s the one used to terminate C strings. While it’s probably not often that you’ll want an embedded NULL in a string, this counting really gets in the way.

Turbo Pascal’s designers compromised on the length of character strings. They figured that, most of the time, character strings did not need to be any longer than 255 characters. That is, after all, a pretty generous length for a line of text. They also figured that if they extended the definition of an array of char a little bit, they could use array element [0] to store the string length, and elements [1..255] for the characters themselves. If each string operation adjusted the length byte appropriately, string operations would be made a lot more efficient, because all that counting would be avoided.

But Windows was coded in C. A large percentage of the Windows API involves passing the addresses of C-type strings. So what were the BPW designers to do?

As I see it, they had three choices:



•	Create a new, Turbo Pascal-like layer between BPW and the Windows API. This layer would convert Pascal strings to C strings and back, as needed, but would slow things down a little. It would also alter the calling sequences, confusing experienced Windows programmers.



•	Throw out the Pascal strings and let the compiler create C-type strings instead. This would mean rewriting the Pascal string library, and it would prevent using any existing assembler code designed for Pascal strings.



•	Invent a new data type for C-type strings, and provide routines for conversion between the old Pascal strings and the new C-type strings.



This last choice is the one the Turbo Pascal for Windows designers made. Called the pChar, it is the data type that Borland Pascal supports for dealing with the Windows API. But we don’t have to be stuck with their decision. Besides, what better way can we become comfortable with the mix of Pascal and C-type strings than by creating a new class of string that combines them both?

Object Strings to the Rescue�tc "Object Strings to the Rescue"�

The ObjectString class—OString for short—provides the link between Pascal strings and C strings, and formalizes within a class the various methods for dealing with those strings.

Perhaps, in a perfect world, you would see the following features in a BPW string class:



•	Strings could be of arbitrary length.



•	Their lengths would be maintained so we wouldn’t have to constantly count characters. 



•	They could be passed to the Windows API without first having to be copied into the C format.



•	They could be implicitly converted to the Pascal format when needed.



•	They could be created as easily as Pascal strings by simply putting some characters within single quotes.



•	Their methods could be implemented mostly by referencing the existing Pascal and pChar string functions.



•	Strings could be stored in as little space as possible, while allowing the programmer to specify additional space if needed for buffering.



Of course, it’s not a perfect world, and we might have to compromise. The “arbitrary length” requirement conflicts with the “convert to Pascal” requirement. No string greater than 255 characters in length can be represented as a Pascal string; the character count won’t fit in the byte assigned to that purpose.

If we decide that the “arbitrary length” requirement is more important, then we’ll lose the chance to implement our string class methods with help from the Turbo Pascal string library.

But the feature you’re least likely to see is the one in which you create your strings as easily as you create Pascal strings. Pascal strings (and, now, pChars) are part of the language, understood by the compiler. No class we invent can have that distinction.

Creating a Unit Skeleton�tc "Creating a Unit Skeleton"�

If you are going to be creating a lot of units, it would be helpful to have a skeleton you could copy.

As I said in Chapter 1, a BPW program is composed of one program file and, probably, several units. There may even be a library. The source code for programs, units, and libraries are kept in files with .PAS extensions; when you compile, BPW knows which kind of file it is (from the keyword Program, Unit, or Library) and produces a file with an .EXE, .TPW, or .DLL extension, respectively.

You’ll almost always want to put each object class in its own unit, so you’ll be creating lots of units. Why not start out with a simple unit skeleton, such as the following:



Unit UnitSkel;



	(***********************************************)

						Interface

	(***********************************************)



	Uses

		WinTypes,

		WinProcs,

		Objects,

		OWindows,

		WinDos;



	(***********************************************)

					Implementation

	(***********************************************)



	End.



You should create a SKELS subdirectory where you keep all your skeleton code. If you type this in, you should name it UNITSKEL.PAS. If you have the optional code disk, you can copy this file.

Besides a unit name and the required keywords Interface and Implementation, the unit skeleton includes a Uses clause, which contains the most often referenced units. There’s no harm in leaving in a reference that your new unit doesn’t actually use, because the BPW linker omits unreferenced code. But, for clarity’s sake, you might choose to remove an unneeded reference from the list. I include all the most-often used ones in the skeleton and remove what I don’t want, on the theory that it’s easier to delete than to type.

WObjects Unit Has Been Replaced

In Turbo Pascal for Windows, the unit that contains the basic behavior of objects is called WObjects. This unit was split into four components in Borland Pascal. The Objects component is shared by all targets, DOS and Windows alike. The other components are OMemory, OWindows, and ODialog. 





The Interface section of a unit is the part that other modules can “see.” It is analogous to a header file in C. This is where you will put the description of the class you intend to share with other program components, as well as any supporting constants and other types, like enumeration values.

The Implementation section is where the actual code will go. Although other modules get to use this code, they cannot actually see it. Any constants you put in the Implementation section, or any procedures or functions you put here that do not have forward declarations in the Interface section, cannot be directly called by another module. In fact, unless you specify otherwise, the compiler will make the machine code prologue for procedures and functions that do not have forward declarations “near”—that is, they cannot  be called from outside this code segment, even if you somehow send their address outside the unit.

We’ll now create a new unit for our tOString class. Use File Manager to create an OSTRING directory, then drag UNITSKEL.PAS to it while holding down the Control key. (That guarantees a copy, not a move, even if both directories are on the same drive.) Select the copy, then rename it OSTRING.PAS using the File Rename command. Finally, double-click on the file; Turbo Pascal for Windows should start and load OSTRING.PAS for you.

How to Create an Object in Seven Steps

Figure 2.1 shows you the seven basic steps to create an object. You can use this as a road map to create your own objects.



Descending from tObject�tc "Descending from tObject"�

The tObject class defines basic behavior for any useful object in the BPW environment. Now that we have a unit skeleton, we’ll learn how to customize it for a class that descends from tObject. 

Practically every object class defined in OWL descends from tObject. Classes that you create should also, in most cases, descend from tObject (if not from a higher-level class), as well. Your first task is to change



Unit UnitSkel;



to



Unit OString;



We’ll now take an initial stab at a tOString definition. Remember, the definition is in the form of a Type statement and goes in the Interface section, since this is not a “private” object class that will only be used internally by other code in this unit.



Unit OString;



	(***********************************************)

						Interface

	(***********************************************)



	Uses

		WinTypes,

		WinProcs,

		Objects,

		OWindows,

		WinDos,

		Strings;



	Type

		pOString = ^tOString;

		tOString = Object (tObject)

			End;



	(***********************************************)

					Implementation

	(***********************************************)



	End.



It’s always easiest to start with the parts you’re sure of, and I’m sure I want tOString to be descended from tObject. Why? Because tObject descendants can be made streamable. That is, they can be written to and read from a file of objects along with other BPW objects. We’ll arrange for that shortly.

Note that I added Strings to the Uses clause. This is a unit that comes with BPW. It contains procedures and functions that deal specifically with the pChar data type. Since we intend to supply conversions to and from pChar, we’ll need this unit’s help.

Limit Unit Names to Eight Characters

Unit names, like other Pascal names, can be of any length, though the first 63 characters must be unique. However, most of the time your unit names must be no more than eight characters long. That’s because when you try to use one of these units, the compiler uses the unit name to construct a filename, and uses that filename to search for it. The exception is in the main library of units (TPW.TPL), where the System, WinTypes, and WinProcs units are stored. If you use TPUMOVER to place a custom unit into the unit library, you can give it a longer name. BPW looks in the library first, and accesses to the library are faster than accesses to TPUs stored conventionally. But it’s a hassle to get a module into that library, and when you have a copy both in the library and out of it, it’s confusing as to which one is being accessed.



Stack versus Heap Allocation�tc "Stack versus Heap Allocation"�

Instances of object classes, like any other variables, can be created in your application’s static data segment, heap memory, or on its stack. Here’s the difference.

Now that we’ve defined our tOString class, it will inherit the tObject’s Init constructor and Done destructor. So even without any methods of its own, with just the type described above, you can create and destroy instances of this totally useless class. You can allocate them directly:



Procedure MyProc;

	Var

		MyOString: tOString;

	Begin

	MyOString.Init;

	MyOString.Done;

	End;



Or you can allocate them on the heap:



Procedure MyProc;

	Var

		MyOString: pOString;

	Begin

	MyOString := New (pOString, Init);

	Dispose (MyOString, Done);

	End;



Any program requires computer memory in which to run. Memory, of course, is built into the computer as a mass of eight-bit bytes. This memory is used to store the program’s code, constants, and data. In a Windows application, these components are kept in separate blocks of memory called segments. 

Memory is addressed via a two-part pointer. The first part contains a segment ID; the second part contains the offset into that segment, at which the desired byte of memory can be found.

When you “directly” allocate any variable—object or not—you are telling the compiler to do the memory management chore for you. You might, for example, include a Var declaration in the area of your program that is not part of any procedure or function, like this:



Program MyProg;

	Uses

		WinCRT;

	Var

		MyObject: tObject;

	Begin

	MyObject.Init;

	MyObject.Done;

	End;



The compiler allocates space for such variables in the data segment. It will exist, and be accessible to your program, as long as the program is running.

Alternatively, you could use the following construct:



Program MyProg;

	Uses

		WinCRT;

	Var

		MyObject: pObject;

	Begin

	MyObject := New (pObject, Init);

	Dispose (MyObject, Done);

	End;



While the end result is identical—a program that does nothing, quickly—the path it takes to get there is quite different. In this case, the only component of the data segment is a four-byte pointer. Space is not allocated for a tObject object at all, at least not at compile time. Not until runtime will MyObject  come into existence, when your program invokes the New function, telling it to allocate the space needed for the kind of object a pObject  points to. It then invokes its Init  constructor, and returns its address. Likewise, the Dispose  procedure will invoke the object’s Done destructor and de-allocate the proper amount of memory, starting at the address contained in the MyObject  pointer.

Similarly, you can place either direct or pointer objects following the Var keyword of a procedure or function. In that case, the objects or pointers will be on the stack, an area of memory that is automatically allocated when the procedure is entered, and de-allocated when it is exited. If you create a tObject, memory for the whole object will be allocated; if you create a pObject, only space for the pointer will appear on the stack and you’ll have to use the New function to allocate the rest of it, as before.

Where does New put the rest of the object? The short answer is the heap, but where is that? Actually, the stack and the heap share the same segment. The idea is that, since allocation of both the stack and the heap are dynamic, they probably both won’t be at their greatest size at the same time. Thus, by letting the stack and heap share the same segment, we can get the most use out of limited memory resource. It’s an old scheme that we seem to be stuck with, even though memory—especially in a computer configured adequately to run Windows 3.1—is not nearly so scarce. 

Stack was created in answer to the problem of procedures and functions: how can the computer efficiently keep track of where execution is to continue when a called procedure exits, and where can local variables be kept? Stack’s solution was to reserve an area of memory that operated on the basis of first-in, last-out. Items could be “pushed” onto the stack, and “popped” off in reverse order, most recent item first. The stack starts at the highest possible memory address in its segment. As items are pushed onto it, the stack pointer contents are automatically decremented. Each time an item is popped from the stack, the stack pointer contents are automatically incremented. Thus, stack management is taken care of by the hardware, with no programmer or compiler intervention beyond specifying the procedure parameters and local variables. The stack grows downward, reaching its greatest volume as the procedures and functions of your program are most deeply nested, and shrinking back to nothing as your procedures and functions exit and your program comes to a close.

The heap is completely different and traditionally shares space in the stack segment in the same way that mice and humans might share a home. Each is oblivious to the other, as long as neither intrudes unexpectedly on the other’s space.

There is no hardware support for the heap; none. The idea is that, just as variables with a lifetime equal to that of a procedure (local variables) are a natural need of programs, so are variables whose lifetimes are not synchronized with procedure entry and exit. For example, you know you need a string to represent the name of the currently opened file, but (at compile time) you don’t know how long the string will be. You don’t want to allocate the maximum possible size at compile time; a few decisions like that and your program’s memory needs will become bloated as an overcooked hot dog. But by making the pointer permanent, and allocating space on the heap (using New or StrNew) for the data, you take only what is needed. When another file is opened, the old filename can be disposed of and a new one allocated.

Understanding the Local and Global Heaps

Experienced Windows programmers know about two heaps: the Local heap, which shares storage with static data and the stack, and the Global heap, which cannot be divided into more than 8192 chunks system-wide. When New or GetMem is used to request memory, the requests are always satisfied from the Global heap. To avoid hitting the 8192-chunk limit, small requests are satisfied from a single chunk, which the BPW Heap Manager controls. Larger requests automatically cause a single Global chunk to be allocated. “Small” is defined as any request equal to or less than the value of the HeapLimit system variable. This value defaults to 1024, but you can change it if you want.

When the BPW Heap Manager allocates a Global chunk for subdividing, the chunk size is equal to the value of the HeapBlock system variable. By default, this value is 8192, but you can change it, too. The BPW documentation recommends that, whatever values you choose, HeapBlock should be at least four times HeapLimit.

Now that Windows no longer supports Real Mode, and most copies of Windows are run on 80386 processors or better (with virtual memory), there is no reason to be so stingy with memory. “System Resources,” as shown by the Program Manager About box, shows me that I always run out of global descriptors (the 8192-chunk limit) long before I run out of memory. So, for any application that is going to actually do any work, I recommend starting values of 8192 for HeapLimit, and 65535 for HeapBlock. You can make HeapBlock smaller if you’re sure you won’t need that much room.





There are times when a direct or stack allocation is appropriate, and times when the heap would be better. What you don’t want to do is pick one and stick mindlessly with it. Each has its place:



•	Use the data segment for an object that is needed for the life of the program and whose size is basically known at compile time.



•	Use the stack for an object that is needed for the life of a procedure or function, and whose size is basically known at compile time.



•	Use the heap for other objects.

Using Constructors and Destructors�tc "Using Constructors and Destructors"�

Objects must be created via a special method called a constructor; eventually they must be destroyed by a destructor method. If you don’t write your own, you can inherit default constructor and destructor methods from your ancestor class (such as tObject). You should know what they do, why they are needed, and what you should put in yours.

In our tOString class usage, we invoked the Init constructor to initialize the object, and the Done destructor to clean it up. But we never even defined these methods, so where did they come from?

This question hits on the first advantage of object-oriented programming. The fragment “ = Object (tObject)” means that our new object class (the tOString) is an object that descends from the tObject class. By “descends,” we mean that the new object will inherit any properties and methods from its ancestor. Actually, a tObject doesn’t have many methods, and no properties; but among its methods are do-nothing Init and Done.

Even though tObject.Init does nothing, you still have to invoke it (or, at least, some constructor for the object you are creating). That’s because invoking a constructor tells BPW to initialize the Virtual Method Table (VMT) for that object. We’ll talk about the VMT later; for now, remember that the surest ways to crash your program—and maybe your Windows session—is to try and access an object you forgot to Init, or to terminate your program without invoking Done.

Also, you must invoke New to obtain a pointer address for an object you intend to store on the heap, and must Dispose of it when you no longer need it. Init and Done, in that case, are part of the syntax of the New/Dispose procedures and thus harder to forget.

Adding Properties for String Storage�tc "Adding Properties for String Storage"�

At their simplest, objects combine the attributes of code and data. The code is implemented as methods; the data is represented by properties. An object does not need to have both, but when it does,  each can be optimized for the other, making the object more powerful.

We still don’t have an object that can actually do anything. So let’s add a few properties for storing the string.

First, we want to store the string in such a way that conversion to and from C-style strings will not require the characters in the string to be copied. We can easily do this by reserving room at the end of the string for a null character.

I am going to take the tack that storing strings longer than 255 characters is more important than conversion to Pascal strings, but you’ll still want to maintain a length value. If we store the length as a Word variable, we’ll be able to store up to 65535 characters. However, we certainly don’t want to allocate space for a 65535 character string every time! So we’ll just keep a pointer to a string and allocate, from the heap, what we need, when we need it.

Note that whether we allocate room for this string on the heap or not has absolutely nothing to do with where the object itself is allocated. The object may be in the data segment, the stack space, or the heap; the characters of the string will still come from the heap.

So we have the following definition:



Type

	pOString = ^tOString;

	tOString = Object (tObject)

		MaxLength: Word;

		Length: Word;

		CString: pChar;

		End;



We are immediately in trouble. If someone allocates space for this object on the stack or from the heap, there’s no telling what might lie in the MaxLength, Length, and CString properties, though we do know it probably won’t be useful. Therefore, we need to initialize these values. And, though the CString property might be initialized to Nil, it’s not likely to stay that way. We’ll need a Done destructor to dispose of any C string we might later allocate.

Finally, the Init method will need to allocate space for the storage of the string, an operation that will be needed many other times as well. If you simply assign a value to MaxLength, the space won’t be allocated. There is no way to automatically link assigning (or retrieving) a value with performing an action in Pascal, as there is in C++. However, we can simulate this by:



•	Making MaxLength “private” so we can’t inadvertently assign a value directly



•	Provide GetMaxLength and SetMaxLength methods to do the job instead.



The Private keyword tells the compiler that the properties and methods that follow are not to be used outside of the class being defined. Public, on the other hand, causes the scope of properties and methods defined afterwards to revert to normal. Since each keyword defines the beginning of a block, they are not terminated with a semicolon.

Therefore, by making SetMaxLength a stand-alone method that Init �can invoke, we avoid the repetition of code.



Type

	pOString = ^tOString;

	tOString = Object (tObject)

		Private

		MaxLength: Word;

		Public

		Length: Word;

		CString: pChar;

		Constructor Init (aMaxLength: Word);

		Destructor Done; Virtual;

		Procedure SetMaxLength (aMaxLength: Word);

		Function GetMaxLength: Word;

		Function GetBufferLength: Word;

		End;

The Public Keyword Was Introduced with Borland Pascal with Objects Version 7.0

If you are still using TPW, you have no easy way to enforce the hiding of properties of methods. The best way to hide properties of objects that others will use may be to simply not document them.



We’ve provided a parameter, aMaxLength, to the Init constructor to help it do its job. And I’m sure you noticed the keyword Virtual after the declaration for Done, but don’t worry about it now. It has to do with the Virtual Method Table I mentioned earlier. Meanwhile, let’s check out the code for the Init constructor. This goes in the Implementation section:



Constructor tOString.Init (aMaxLength: Word);

	Begin

	Inherited Init;

	MaxLength := 0;

	Length := 0;

	CString := Nil;

	SetMaxLength (aMaxLength);

	End;



Since this is our first constructor, we’ll take it line by line.

The first thing you may notice is that the constructor header is not identical to the one in the type statement: here we’ve fully qualified the method name by prefixing it with the name of the object class to which it belongs. This will avoid confusing both the compiler and ourselves, and allow us to place more than one class description with the same method name in the same unit.

Contrary to what you might think, the Begin keyword can generate code. Since tOString contains at least one virtual method (the Done destructor, inherited from tObject), there is a VMT to be initialized. If you look at the machine code that is produced (using Turbo Debugger’s CPU window, for example), you’ll see several instructions perched after Begin and before the call to the ancestor Init constructor.

After Begin, we invoke tObject’s Init constructor using the Inherited keyword. This keyword automatically invokes the specified method of the class’ immediate parent. tObject  is our ancestor class, so tObject.Init is the constructor we’ll get. We could ask for it that way, but by using the Inherited keyword, we are more flexible. We can change our minds later and derive our class from some other ancestor without having to retype all our references to parental methods.

The Inherited Keyword Was Introduced in Borland Pascal with Objects Version 7.0

If you are using TPW, you’ll have to fully qualify the ancestor method you wish to invoke.





But I’ve already mentioned that tObject.Init  is a do-nothing method. So why bother invoking it?

Anytime we create a new class that descends from an existing class, we have to decide just how much of the ancestor we intend to keep, and how much we intend to change. If we kept it all and didn’t add anything, we would have a new class that was identical to the old class.

So, typically, the descendent class will have a number of methods with names identical to methods of the ancestor, and perhaps some additional methods as well. When a method shares a name with an ancestor method, we say the new method overrides the ancestor method. We must then choose whether the new method will replace the old method, or augment it. You replace the old behavior by not invoking the old method from the new method. You augment it by invoking the ancestor method, then doing whatever else you wanted to do. (You can remove an ancestor behavior by supplying an overriding method that does nothing at all.)

If your class is derived from any class but tObject, you must always invoke your ancestor’s Init constructor, because it is needed to initialize whatever properties the ancestor might have. If you have no properties of your own, you won’t need your own Init, either.

We know that tObject has no properties, so we could safely omit its �Init  from ours, but that may be a poor idea. First of all, doing so relies on your knowledge of another class’s internals, and this is exactly what an object-oriented programmer is supposed to avoid. Secondly, a future version of BPW may enhance the tObject class so that its Init method does do something. Finally, it makes for a good habit: if you’re descending a class based on one of the other OWL classes and forget to invoke its constructor, your program will crash. If you always include it when putting together an Init method, you’ll be less likely to forget.

The next two lines of the Init code initialize MaxLength and Length to zero. All properties, whether they are objects or not, must be initialized during execution of the constructor. This is because you have no idea what might lie in the bytes of memory allocated to your object. Object properties must be initialized by their own constructors, of course; scalar variables just need default values assigned. So we assign Nil to CString, to indicate that no memory has been allocated for the string yet. Nil is a special address that says “this pointer does not point to anything right now.” Nil is, in fact, a pointer with a value of zero for the segment and offset. Since, under protected mode, this value is invalid, there’s no mistaking Nil for a legitimate address.

Finally, we invoke SetMaxLength, because there’s more involved in setting MaxLength than merely assigning the value. MaxLength is intended to reflect how much memory has actually been allocated for the string. Setting MaxLength, then, means freeing any space that may have been previously allocated, and allocating a fresh block of the new size. (We previously set MaxLength to zero to ensure that SetMaxLength �would work correctly.

After the call to SetMaxLength, our construction chores are completed.

Following tradition, the next code in the unit is the Done destructor. But first let’s peek at SetMaxLength:



Procedure tOString.SetMaxLength (aMaxLength: Word);

	Begin

	If MaxLength <> aMaxLength then

		Begin

		If Assigned (CString) then

			FreeMem (CString, MaxLength + 1);

		MaxLength := aMaxLength;

		If MaxLength = 0 then

			CString := Nil

		else

			GetMem (CString, MaxLength + 1);

		End;

	If Assigned (CString) then

		CString[0] := #0;

	Length := 0;

	End;



If it turns out that MaxLength isn’t actually being changed, we don’t have to do anything but truncate the stored string to a zero length.  But if the new and original values differ, we have work to do. CString might be Nil or memory might have been allocated. If memory was allocated, it is freed. The assignment to MaxLength is then made. If MaxLength has been set to zero, the tOString object will have no memory allocated to the string, in which case CString will be set to Nil. Otherwise, we must allocate space to it—enough for MaxLength characters and a terminating null. (Unlike most C functions, tOString will not require programmers to be constantly aware of that null.) Finally, calling SetMaxLength will clear the string. So, if CString points to allocated memory, we place that terminating null in the first character position, and set Length to zero. GetMaxLength and GetBufferLength are trivial:



Function tOString.GetMaxLength: Word;

	Begin

	GetMaxLength := MaxLength;

	End;



Function tOString.GetBufferLength: Word;

	Begin

	GetBufferLength := MaxLength + 1;

	End;



GetMaxLength is used in lieu of directly accessing the MaxLength property. Many Windows API calls require the length of a buffer including space for a terminating null; that’s what GetBufferLength is for.

If MaxLength weren’t private, you wouldn’t even need these methods. Still, this type of indirection is considered good form in OOP circles. It gives you the freedom, later, to change the way you store or calculate MaxLength without changing this method, or the other methods that call it.

Done is also simple. If space has been allocated to CString, it is freed. Then, just as the ancestor class’s Init must be invoked from our constructor, so must the ancestor Done be invoked here:



Destructor tOString.Done;

	Begin

	If Assigned (CString) then

		FreeMem (CString, MaxLength + 1);

	Inherited Done;

	End;

Creating Methods�tc "Creating Methods"�

Writing an object method is like writing a procedure or function… almost. Each has an invisible, assumed “Self” parameter, because an object’s methods are intended to work on itself.

Of course, our tOString object would be a lot more useful if we could actually assign a string to it. Moreover, we need ways to get both Pascal and C strings into it. It would also make sense to be able to get the text from another tOString.

We could write three completely different methods, one for each input string type. However, the work of allocating space and copying the string is the same for each. So why not build a generic method that actually does the work, with three conversion methods to invoke it?

My personal technique for identifying underlying methods I use to provide base functionality for higher-level methods is to append an under�score to the method name. Here’s the base method for assigning text:



Procedure tOString.SetText_

		(aText: pChar; aLength: Word);

	Begin

	If aLength > MaxLength then

		SetMaxLength (aLength);

	Length := aLength;

	Move (aText[0], CString[0], Length);

	CString[Length] := #0;            

	End;



In this method, aText  is a simple array of characters and aLength is the number of characters it contains. Only if this length is greater than MaxLength must we re-allocate the memory assigned to CString, which we do by invoking SetMaxLength—another reuse of code. The Borland Pascal standard function Move copies the characters into the assigned storage, and the last line adds the null terminator to the correct byte.

Given this worker method, implementing the three string type-specific methods is trivial:



Procedure tOString.SetTextC (C: pChar);

	Begin

	If not Assigned (C) then

		Length := 0

	else

		SetText_ (C, StrLen (C));

	End;



Procedure tOString.SetTextP (const P: String);

	Begin

	SetText_ (@P[1], System.Length (P));

	End;



Procedure tOString.SetText (Var O: tOString);

	Begin

	SetText_ (O.CString, O.Length);

	End;



Each string simply determines the address of the first character to be copied, according to the input data type, and the number of characters. (Zero-length C strings are often represented by a Nil pointer, so we accommodate that.) By passing this information on to SetText_, the work gets done with a minimum of excess code. As a side benefit, SetText_ can be invoked from outside this unit after all. That may not have been its intended purpose, but there’s no harm in it.

Having gotten the string into a tOString, we now have to be able to get it out. As with input, we need to access our tOstring in Pascal, C, or OString format. The C format is directly accessible through the tOString’s CString property, so we don’t have to worry any further about that. For Pascal and OStrings, we’ll need to create methods.

The method that returns a Pascal-style string is entirely accomplished by the StrPas  function of the Strings unit. This function is so efficient that it cannot be improved. Still, for the sake of a clean interface, I encapsulate it as the PString method:



Function tOString.PString: String;

	Begin

	PString := Strings.StrPas (CString);

	End;



The next method, which produces a duplicate tOString, simply allocates a new instance of the class on the heap and invokes the SetTextO method we’ve already written. It then returns the address of the new object:



Function tOString.OString: pOString;

	Var

		Result: pOString;

	Begin

	Result := New (pOString, Init);

	Result^.SetText (Self);

	OString := Result;

	End;



Even though OString is not itself a constructor, it invokes  a constructor. Thus, the pointer OString returns will need to be disposed of eventually.

Shortcut Constructors�tc "Shortcut Constructors"�

There’s no rule that says you must have just one constructor, or that its name must be Init, for that matter. Init is a convention, albeit a useful one, that says “create an empty or minimally initialized instance of this class.” But we can also provide less minimal initializations.

There is one of our design constraints that we have thus far ignored: the one that wanted tOString variables to be as easy to create as native Pascal variables. 

As stated earlier, this is not a goal we can ever meet. But we can simplify the creation of a tOString. The key is to have more than one constructor. Probably 99 percent of the time we create a tOString we’ll immediately assign a string to it. Why not combine those two operations, then, into a single constructor method? Here they are:



Constructor tOString.InitTextC (C: pChar);

	Begin

	tOString.Init (0);

	tOString.SetTextC (C);

	End;



Constructor tOString.InitTextP (P: String);

	Begin

	tOString.Init (0);

	tOString.SetTextP (P);

	End;



Constructor tOString.InitText (Var O: tOString);

	Begin

	tOString.Init (0);

	tOString.SetText (O);

	End;

Concatenating Strings�tc "Concatenating Strings"�

Now we need to give our OString class the ability to actually do something useful. This is not a book on basic programming so let’s not pick anything challenging like language parsing or masked formatting. Here’s something simple, basic, and useful: string concatenation. From the point of view of a given tOString object, this is an append operation.

Borland Pascal has a great syntax for concatenation: simply unite the two components with the “+” operator and assign the value to a new Pascal string variable.

Unfortunately, BPW does not provide a facility like operator overloading in C++, where you can provide your own replacement functions for operators. We’ll have to do our concatenation within a conventionally-invoked method, which, generically, we name Append in honor of the relationship of the operation to the object itself. As in string assignment, we’ll need a matched set of three Append  methods: one each for C-style strings, Pascal strings, and tOStrings. And, again, as in SetText, the most efficient technique is to put the common code in an “internal” method, Append_, which is invoked from the three conversion methods. Here’s Append_:



Procedure tOString.Append_

		(More: pChar; MoreLength: Word);

	Var

		Temp: pChar;

		TempLength: Word;

	Begin

	If Length = 0 then

		Begin

		SetText_ (More, MoreLength);

		Exit;

		End;

	If (Length + MoreLength) > MaxLength then

		Begin

		Temp := StrNew (CString);

		TempLength := Length;

		SetMaxLength (Length + MoreLength);

		StrCopy (CString, Temp);

		Length := TempLength;

		StrDispose (Temp);

		End;

	Move (More[0], CString[Length], MoreLength);

	Length := Length + MoreLength;

	CString[Length] := #0;

	End;



If the current length of the string is zero, this is really the same as a SetText_ operation; so we invoke that method and then exit. Otherwise, if the space we allocated is not adequate to contain the old string plus the appended string, we must re-allocate CString, or we would undoubtedly tramp on some other piece of data or even cause a General Protection Fault. But we don’t want to lose the old string, and SetTextLength will clear it. So we create a temporary copy of the old string using StrNew, change the MaxLength, copy the temporary string back, and release it. This may seem like a lot of data manipulation, but remember, it will only happen if MaxLength is too small for the combination string. Think of it as a safety valve. If you know a given string will be getting bigger, just allocate a generous amount of memory to it to start with and this copying code will never be executed.

Confident now that we won’t cause a General Protection Fault by trampling outside our assigned memory block, we can Move the characters to be appended to the end of CString and add the new terminating null.

As in the case of the SetText methods, the AppendC, AppendP, and Append methods now boil down to one-liners:



Procedure tOString.AppendC (MoreC: pChar);

	Begin

	Append_ (MoreC, StrLen (MoreC));

	End;



Procedure tOString.AppendP (MoreP: String);

	Begin

	Append_ (@MoreP[1], System.Length (MoreP));

	End;



Procedure tOString.Append (Var More: tOString);

	Begin

	Append_ (More.CString, More.Length);

	End;

Beyond Compare �tc "Beyond Compare "�

Probably the most common string activity is string comparison. Depending on the circumstances, comparisons might be case sensitive or insensitive for the entire length of the string or for a portion of it, and a true/false operation or one that must determine which of two strings should be placed lexically ahead of the other. As a complete solution to string handling, tOString must supply these abilities in a quick and easy-to-use manner.

The first design consideration that arises is how to implement all these features in a way you can use later without having to refer back to your code. To help answer this, we can look at how they’ve been implemented in other string-management systems.

In C, for example, there are separate functions for each variant: strcmp compares case-sensitive strings for their full length, stricmp compares case-insensitive strings, strncmp compares strings up to a maximum length, strnicmp compares case-insensitive strings up to a maximum length, and so on. If we were to use this technique we could provide clearer names, but there would still be a lot of methods. BPW’s STRINGS unit mimics the C plethora of string functions for pChar (C-style) strings.

In Borland Pascal, the handling of BP-style strings is done by the compiler. String comparison is handled by the standard operators: =, <, >, and <>. Unfortunately, as designers of object methods, that technique is not available to us.

Surprisingly, we may find the answer in—of all places—80x86 Assembler. For any processor, there are a limited number of separate instructions available, but each has many variants. On the 80x86 family of CPUs, there is a single string compare instruction (two if you count byte compares and word compares separately), which the programmer fine-tunes by placing values in several registers. One controls the direction of the compare (front to back or vice versa), another controls the number of bytes to be compared, and so on. Why can’t we do that, substituting properties or parameters for registers?

In trying to decide whether to control operations via properties versus parameters, you should ask yourself: will I need to supply different options with every call, or could I just set them once and forget it? In other words, is the option more associated with the object, or with the operation?

Case insensitivity, for example, tends to be an attribute of a string object. This is not the same as always storing the characters in uppercase, or lowercase, for that matter. It does mean that the object should be smart enough, when a compare operation is taking place, to compare same-case copies of the strings instead of the originals.

On the other hand, the number of characters to be compared—all of them, or just a few—is more an attribute of the operation; it makes more sense to pass the compare count as a parameter.

Finally, it seems intuitive that a “match” operation and a more general “compare” used for ordering purposes would be separate methods. But this is understating the case, and pointing out a deadly tendency when entering the world of object-oriented programming. When you perform a comparison of any two objects, it is you performing the comparison—not the objects. A method, on the other hand, is something that an object “knows” how to do to itself. Therefore, we must have three sets of methods: isBefore, Matches, and isAfter. Each method will be implemented as a Boolean function.

With the foregoing in mind, let’s get started. First, we add a CaseSensitive property to tOString:



Type

	pOString = ^tOString;

	tOString = Object (tObject)

			

			

		CaseSensitive: Boolean;

		Constructor Init;

			

			



It must be initialized to True, since most strings are case sensitive. (If you had a need for a class of case-insensitive strings, you could always derive a new class and override the constructor.)



Constructor tOString.Init;

	Begin

			

			

	End;



To change the case sensitivity of a given string, simply assign the desired value to the property:



MyString.CaseSensitive := False;



Although, from the point of view of a tOString object, the operations are isBefore, Matches, and isAfter, a comparison still must be made. If each of the nine methods—three kinds of comparisons on three flavors of string—deals with the details of the string format, the actual comparison code can be the same for all. Note that the following function is not a method; it is a “helper function,” residing in the Implementation section of the OSTRING unit and therefore available only to the tOString methods that need it. The same is true of the function’s return type, tOrders: 



Type

	tOrders =

		( A_Before_B, A_Matches_B, A_After_B );



Function Order

		(

		A, B: pChar;

		Length_A, Length_B: Integer;

		CaseSensitive: Boolean

		): tOrders;

	Var

		CompareCount: Integer;

		Result: Integer;

	Begin

	If Length_A = 0 then

		Begin

		If Length_B = 0 then

			Order := A_Matches_B

		else

			Order := A_Before_B;

		End

	else if Length_B = 0 then

		Begin

		If Length_A = 0 then

			Order := A_Matches_B

		else

			Order := A_After_B;

		End

	else

		Begin

		If Length_A < Length_B then

			CompareCount := Length_A

		else

			CompareCount := Length_B;

		If CaseSensitive then

			Result := StrLComp (A, B, CompareCount)

		else

			Result := StrLIComp (A, B, CompareCount);

		If Result < 0 then

			Order := A_Before_B

		else If Result = 0 then

			If Length_A < Length_B then

				Order := A_Before_B

			else if Length_A = Length_B then

				Order := A_Matches_B

			else

				Order := A_After_B

		else

			Order := A_After_B;

		End;

	End;



tOrders exists to document the possible return values. Compare it to the sloppy return value of the C-style string compare functions, where “less than zero” means the first parameter comes before the second and “greater than zero” means the opposite.

The Order function has four small sections of code. The first determines whether either of the string lengths is zero; if one is, the return value can be calculated directly. (This avoids invoking one of the string compare functions with a Nil pointer.) If neither string length is zero, we calculate the compare count, which must be the lesser of the string lengths. Even though A and B, the two strings being compared, are defined as pChar types, this is more for convenience. There is no guarantee that there will be a terminating null character. In fact, if a Pascal-style string is being compared, there won’t be. The second section invokes either StrLComp or StrLIComp, depending whether the comparison is to be case sensitive or not. Finally, based on the return from StrLComp or StrLIComp, the function determines whether A comes before, after, or matches B.

With the Order function available to the tOString methods, they take longer to type than to figure out. The isBefore methods look like this:



Function tOString.isBefore

		(Var Test: tOString): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		Test.CString,

		Length,

		Test.Length,

		CaseSensitive and Test.CaseSensitive);

	isBefore := (Result = A_Before_B);

	End;



Function tOString.isBeforeC

		(TestC: pChar): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		TestC,

		Length,

		StrLen (TestC),

		CaseSensitive);

	isBeforeC := (Result = A_Before_B);

	End;



Function tOString.isBeforeP

		(TestP: String): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		@TestP[1],

		Length,

		System.Length (TestP),

		CaseSensitive);

	isBeforeP := (Result = A_Before_B);

	End;



In the first of the three methods, which compares against another tOString object, if either object is not case sensitive, the operation is performed insensitive to case. The methods that deal with C and Pascal strings use only the object’s own CaseSensitive property to determine how the operation is to be performed, since this is not an attribute of C or Pascal strings. The only other difference between the three methods is the manner of passing the address of the array of characters to be compared.

The Matches methods are nearly identical to the IsBefore methods:



Function tOString.Matches

		(Var Test: tOString): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		Test.CString,

		Length,

		Test.Length,

		CaseSensitive or Test.CaseSensitive);

	Matches := (Result = A_Matches_B); 

	End;



Function tOString.MatchesC (TestC: pChar): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		TestC,

		Length,

		StrLen (TestC),

		CaseSensitive);

	MatchesC := (Result = A_Matches_B); 

	End;



Function tOString.MatchesP

		(TestP: String): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString,

		@TestP[1],

		Length,

		System.Length (TestP),

		CaseSensitive);

	MatchesP := (Result = A_Matches_B); 

	End;



As are the isAfter methods:



Function tOString.isAfter

		(Var Test: tOString): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString, Test.CString,

		Length, Test.Length,

		CaseSensitive or Test.CaseSensitive);

	isAfter := (Result = A_After_B);

	End;



Function tOString.isAfterC (TestC: pChar): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString, TestC,

		Length,

		StrLen (TestC),

		CaseSensitive);

	isAfterC := (Result = A_After_B);

	End;



Function tOString.isAfterP

		(TestP: String): Boolean;

	Var

		Result: tOrders;

	Begin

	Result := Order (CString, @TestP[1],

		Length,

		System.Length (TestP),

		CaseSensitive);

	isAfterP := (Result = A_After_B);

	End;

Gently Down the Stream �tc "Gently Down the Stream "�

Neither Pascal nor Borland Pascal makes an aesthetic provision for a file containing more than one kind of record. Sure, you can use BlockRead and BlockWrite, but that’s dropping down to low-level I/O.

With BPW, any object descended from tObject inherits the ability to “stream,” that is, to be poured into or out of a file containing it and other objects. The ability is latent; you do have to write a few lines of code to bring it out. But the effort is so trivial that, even if you don’t need to stream an object right now, I recommend adding the code anyway. Knowing it’s there may well encourage you to reuse the class later, even if on a different project.

To make a class streamable, do the following:



•	Make it descend from tObject



•	Create a Load constructor and a Store method



•	Register the class



In the case of tOString, we’ve already made it descend from tObject. The Load and Store methods have only to invoke the corresponding methods of their ancestor, and add to it the loading or storing of the properties that belong uniquely to this class. To load an object, you have to use tStream’s Read method; to store one, you use tStream’s Write method. (A tStream object is passed to Load and Store as a parameter.)

Here’s tOString’s Load constructor:



Constructor tOString.Load (Var S: tStream);

	Var

		aMaxLength: Word;

	Begin

	MaxLength := 0;

	Length := 0;

	CString := Nil;

	S.Read (aMaxLength, SizeOf (aMaxLength));

	SetMaxLength (aMaxLength);



	S.Read (Length, SizeOf (Length));

	S.Read (CString^, Length);

	CString[Length] := #0;



	S.Read (CaseSensitive, SizeOf (CaseSensitive));

	End;



If you have descended from any class other than tObject, remember to invoke your ancestor’s Load method before you do any loading of your own. tObject doesn’t have a Load method because it has no properties. 

Even without looking up the calling sequence of the Read method, it should be easy to follow the logic of this constructor. First, we set MaxLength and Length to zero, and CString to Nil. Then, we read in the MaxLength property, but we read it into a local variable. We have to tell Read the number of bytes to transfer from the stream (the SizeOf function provides that nicely) so we can invoke the SetMaxLength method. After all, why rewrite the memory allocation code? 

Next, we obtain the Length property, which tells us how many bytes to read into CString and where to put the terminating null. Finally, we get the value of the CaseSensitive flag.

As you can guess, the Store method is the inverse of the Load constructor:



Procedure tOString.Store (Var S: tStream);

	Begin

	S.Write (MaxLength, SizeOf (MaxLength));

	S.Write (Length, SizeOf (Length));

	S.Write (CString^, Length);

	S.Write (CaseSensitive, SizeOf (CaseSensitive));

	End;



It’s a little simpler because Store is a method, not a destructor; we don’t have to de-allocate CString.

Finally, we must make certain that tOString is “registered” to the stream system. There’s a predefined record type, tStreamRec; we must create a variable of this type and initialize it with appropriate values, then pass it to the RegisterType system function. The initialization can be done as a constant declaration; the best place to register the object is in the unit initialization code area, between the final Begin and End of the unit:



Const

	rOString: tStreamRec =

		(

		ObjType: 2001;

		VmtLink: Ofs (TypeOf (tOString)^);

		Load: @tOString.Load;

		Store: @tOString.Store

		);



Begin

RegisterType (rOString);

End.



By convention, the name of the registration record is the same as the name of the object class, with an “r” instead of a “t.” It doesn’t matter too much where in a unit you put the registration record. We put it in the Implementation section where it’s invisible outside of the unit.

The ObjType field is the only one you have to worry about, because you have to make up that number. OWL has reserved the numbers from 0 through 99, so you have plenty of numbers (up to 65535) to choose from. But they must be unique across your program. I chose 2001 because this is the first class defined in Chapter 2. A better approach for a full-time developer is to keep a dictionary of classes you’ve created, including object type codes. Then, each time you create a new object class, you can just get a number for it from the end of the list.

Register All Classes!

The ObjectWindows Programming Guide says that OWL defines registration records for all the standard object classes, and that there are procedures that register them all. But it does not actually invoke those procedures!

Actually, you must either specifically register any standard objects you intend to stream, or invoke RegisterObjects, RegisterOWindows and RegisterODialogs yourself. If you forget and try to stream, say a tCollection object, your program will die, possibly taking the rest of your Windows session with it.



Using Resources�tc "Using Resources"�

The resources of a Windows applications include bitmaps, dialog box templates, and icons. They also include strings; specifically, the strings which are to be displayed to the user. The primary purpose of this is to make translation of an application into another language relatively painless. If all the display strings are kept in a single stringtable resource, they can be translated and the program relinked, without having to recompile it. The translators don’t even need access to the program’s source code. With the addition of three simple methods, we can make the use of resource strings as easy as any others.

With the foundation presented in the previous sections, the InitTextR, SetTextR, and AppendR methods consist of no more than a call to LoadString, the API function that retrieves strings from the application resource pool, and a call to the appropriate “workhorse” method. The local buffer into which the loaded string is placed temporarily is 256 characters wide. Since the current limit on string resources is 128 characters, this is more than enough for now and even provides some protection against enhancements to future versions of Windows.



Constructor tOString.InitTextR (ID: Word);

	Begin

	tOString.Init (0);

	tOString.SetTextR (ID);

	End;



Procedure tOString.SetTextR (ID: Word);

	Var

		Buffer: Array [0..256] of Char;

		BufLen: Word;

	Begin

	BufLen := LoadString (hInstance,

		ID, Buffer, SizeOf (Buffer));

	SetText_ (Buffer, BufLen);

	End;



Procedure tOString.AppendR (ID: Word);

	Var

		Buffer: Array [0..256] of Char;

		BufLen: Word;

	Begin

	BufLen := LoadString (hInstance,

		ID, Buffer, SizeOf (Buffer));

	Append_ (Buffer, BufLen);

	End;

Window Text�tc "Window Text"�

Every window, even windows with no visible text like gauges or bitmaps, have an associated text property. In overlapped windows, this text property appears as the caption; in a static or edit control it is the text of the control. The ability to obtain this text and copy it into a tOString object will prove to be handy many times.

To get the text from a window, you use the Windows API call GetWindowText. This function requires you to supply a handle to the window whose text you’d like. In programs written directly to the Windows API, all windows are identified by their handles, a 16-bit value. However, in OWL, windows are normally represented as objects of tWindowsObject class (or one of that class’ descendants). The tWindowsObject class has an hWindow property: this is the window handle we’ll need.

Don’t Use hWindow until SetupWindow �Has Been Called

The hWindow property of a tWindowsObject is zero (a null value) until after the SetupWindow method has been invoked. This is because the physical Windows window does not exist until then. In Windows, a window has certain properties (not to be confused with OWL object properties), some of which must be set before the window is created, and others that can only be manipulated afterward. This has been translated into tWindowsObject objects by having the pre-creation work done in the tWindowsObject.Init constructor, and the post-creation work done in the tWindowsObject.SetupWindow method. The default SetupWindow is often augmented in descendant classes.

Therefore, you must be careful not to try and use hWindow before SetupWindow has been called. One way to do this is to invoke the methods that need hWindow within SetupWindow processing itself. Another is to consistently check the hWindow property before using it, and to refrain from using it if it contains a zero.





Another requirement to using GetWindowText is that we supply a buffer into which the text can be placed, and specify the size of the buffer so that GetWindowText won’t accidentally cause a General Protection Fault by trying to send more characters than there is room for. That means we’ll have to pre-allocate the tOString’s CString property. GetWindowText returns the size of the string, but of course by then it’s too late. Fortunately, there is another API function, GetWindowTextLength, which will help us with that problem.



Constructor tOString.InitTextW

		(aWindow: pWindowsObject);

	Begin

	Init (0);

	SetTextW (aWindow);

	End;



Procedure tOString.SetTextW (aWindow: pWindowsObject);

	Var

		RequiredLength: Integer;

	Begin

	RequiredLength :=

		GetWindowTextLength (aWindow^.hWindow);

	If RequiredLength > MaxLength then

		SetMaxLength (RequiredLength);

	Length :=

		GetWindowText (aWindow^.hWindow,

			CString, GetBufferLength);

	End;



I haven’t included an AppendW method because, frankly, I’ve never needed one. I often append to a window caption, but I’ve never had to append a window caption to something else. If you should find yourself needing such a method, though, you should now be able to create one.

Recalculating the Length�tc "Recalculating the Length"�

One of the provisions we’ve made for the tOString class is that of pre-allocating the buffer. This is to support use of this class in passing the address of the buffer to a Windows API function that will place text in it. When text is assigned in this manner, however, the Length property is not incremented. It is up to us to reset it.

Some Windows API functions place text in the supplied buffer and return the number of characters transferred. To use one of these functions, follow this example:



MyString.Length :=

	GetWindowText (hWindow, MyString.CString, MyString.GetBufferLength);



However, some functions do not return a transfer count. In that case, we could invoke the StrLen function each time and assign its result to Length. But, if we were to do that, we’d also have to include the STRINGS unit in our application’s Uses clause. Why not just encapsulate this behavior in a method?



Procedure tOString.RecalcLength;

	Begin

	Length := StrLen (CString);

	End;



Now, after invoking a function that transfers characters but does not return a count, simply invoke the tOString’s RecalcLength method and be done with it.

The WinCRT Unit�tc "The WinCRT Unit"�

Quickie BPW programs can be thrown together using the WinCRT unit. They use standard Borland Pascal syntax and can even use the standard functions WriteLn to send output to the screen and ReadLn to get input from the user. Although the result cannot be truly called a Windows application, it does run in the Windows environment and can be an excellent tool to test code without embedding it in a “real” �application.

The reason that we’ve waited so long to test anything is that this is ObjectWindows, and we have yet to actually write a skeleton for an ObjectWindows application. But we don’t need to worry about that. BPW includes a unit called WinCRT that permits standard Borland Pascal programs to be compiled and run in a window. I don’t normally approve of WinCRT, since the resulting program will not follow Windows application style in any way. But for a quick test of a component, WinCRT has no equal.

Actually, if you are sharp-eyed enough, you might have spotted a Uses keyword with WinCRT earlier in this chapter. I mentioned that such a program would compile and run, but very quickly, because it produced no output. But now that we can concatenate strings, we can actually try out the tOString class.

Your test program should look like this, saved as TEST1.PAS:



Program Test1;



Uses

	WinCRT,

	OString;



Var

	O1, O2: tOString;



Begin

O1.InitTextC ('Hello ');

O2.InitTextP ('OWL Lovers!');

O1.Append (O2);

WriteLn (O1.PString);

O2.Done;

O1.Done;

End.



Select the Run..Run command and the window shown in Figure 2.2 should appear.

�IMPORT F:\\INSIDER\\FINAL\\FIG02-02.TIF \* mergeformat���

Figure 2.2: The test is successful!

In one fell swoop, we’ve tested three constructors (because InitTextC and InitTextP invoke Init), the Append method, and the Done destructor. You should be able to think of simple variations of TEST1 to test the few methods we’ve missed.

Now, we have obviously just begun what can turn out to be a pretty ambitious project: wrapping all our string-handling needs into a single class. But, since this is not a book on basic programming, we’re not going to agonize over the details. Instead, in the appendix, you’ll find a complete OString class that you can use as is, or modify to suit your needs. We will use some of these other methods in future chapters, as special string needs arise, which are best provided for in this class.

Remember how important it is to develop incrementally. If you test after each few lines you add, you’ll be very confident that you’re building on a solid foundation. Keep your method code short, as we did with tOString, and it will always be easy to find the occasional bug before it can do any damage.

If a bug should rear its ugly head, don’t be afraid to use the Run..Debugger command to track it down. This starts the Turbo Debugger and positions it at the beginning of your code. Simply step and trace through until you find the line that causes the error. If you have kept your developmental increments small enough, it should never take you long to determine which of the four or five lines you just added is at fault. In fact, with incremental development, many times you won’t even need the debugger. And keep in mind that with incremental development, it’s hard to make any programming error that doesn’t turn out to be a typo.

The ClasSkel Unit�tc "The ClasSkel Unit"�

Earlier in this chapter, we made a unit skeleton, which will be useful many times. But we’ve seen that a lot of other components go into a class unit. Why not make a skeleton for that, too?

CLASSKEL.PAS is easily constructed by making a copy of OSTRING.PAS and removing all the tOString-specific stuff:



Unit ClasSkel;



	(***********************************************)

						Interface

	(***********************************************)



	Uses

		WinTypes,

		WinProcs,

		Objects,

		OWindows,

		WinDos;



	Type

		pUnitSkel = ^tUnitSkel;

		tUnitSkel = Object (tBaseObject)

			Constructor Init;

			Destructor Done; Virtual;

			Constructor Load (Var S: tStream);

			Procedure Store (Var S: tStream);

			End;



	(***********************************************)

					Implementation

	(***********************************************)



	Constructor tUnitSkel.Init;

		Begin

		Inherited Init;



		End;



	Destructor tUnitSkel.Done;

		Begin



		Inherited Done;

		End;



	Constructor tUnitSkel.Load (Var S: tStream);

		Begin

		Inherited Load (S);

		S.Read (?, SizeOf (?));

		End;



	Procedure tUnitSkel.Store (Var S: tStream);

		Begin

		Inherited Store (S);

		S.Write (?, SizeOf (?));

		End;



	Const

		rUnitSkel: tStreamRec =

			(

			ObjType: ?;

			VmtLink: Ofs (TypeOf (tUnitSkel)^);

			Load: @tUnitSkel.Load;

			Store: @tUnitSkel.Store

			);



	Begin

	RegisterType (rUnitSkel);

	End



Note that the tUnitSkel class is derived from tBaseObject. There is no such class as tBaseObject, but your new classes will almost always be derived from something. This way, a simple Edit..Replace will change tBaseObject into whatever real class is appropriate. I deliberately left question marks where you’ll have to fill in the blanks later. The question marks will result in syntax errors if you forget to fill them in before trying to compile.

Other Methods to Our Madness

Here is a partial list of other tOString methods listed in the appendix and also available on the optional disk, along with a brief note about anything interesting you may find in the code:



GetToken: Given a character to use as a delimiter, this method divides the string into tokens, strips off the first, and returns it as another tOString.



UpCase, Downcase, ProperCase: These methods alter the case of the string. They also exhibit the use of an overlayed, 1-based array pointer to avoid the need to constantly subtract one from the length of the string when looping through its characters.



Encrypt, Decrypt: These methods use a simple substitution table to achieve string encryption. The resulting strings are still composed of printable characters, but are unreadable to anyone who is not a cryptographer.



MatchesW: This method will often be used to determine if the user has made a change to an edit control or not. Sometimes a user will make a change but then revoke it. This method makes it possible to track such shenanigans.



LeftMost, Rightmost, Substring: These methods parallel the Left$, Right$ and Mid$ functions of Basic. Substring is also similar to the Pascal Copy function. In each case, the string affected is the object’s own.


