Windows’ primary strength is not that it is simply a multitasking environment, but that it also provides a framework in which the multiple tasks can work together. The Windows Clipboard is the most-used component of that framework. It lets you transfer information in various formats, and even assists the application in deciding which format should be used.


Building the tClipboard Class�tc "Building the tClipboard Class"�


Encapsulating the behavior of the Windows Clipboard into a class should now seem second nature to you. In the next few sections, we’ll design the framework of the tClipboard class, and then enhance that framework to accommodate the many Clipboard formats that your applications will need to support.


Designing the tClipboard Class�tc "Designing the tClipboard Class"�


Basic Clipboard use usually involves one format—text—and simple cutting or copying to, and pasting from, the Clipboard. Our design must be able to do that and more. 


At its absolute simplest, Clipboard management is trivial. To place some text in the Clipboard, you simply:





1.	Open the Clipboard





2.	Clear its previous contents





3.	Place a copy of the text into the Clipboard





4.	Close the Clipboard





But placing text in the Clipboard, or getting it out, is only a small part of the story. Although the Clipboard is intended to hold just one conceptual item at a time, the fact is it can hold almost unlimited types of items, each a different formatting of the conceptual item. For example, if the user selects the word “Hello” in your application and chooses Edit..Copy, your app will presumably place “Hello” into the Clipboard in the cf_Text format. But it could also place a picture of the word “Hello” in cf_Bitmap format so that it could be pasted into, say, Paintbrush. It could also generate a cf_MetaFilePict (commonly called a picture) of the text that could be sized and imported by Resource Workshop’s Icon Editor.


Again, just placing text in the Clipboard is an extremely simple, straightforward operation. In a traditional (non object-oriented) �environment, you wouldn’t think twice about coding those forty or fifty lines for each application. But we would like to encapsulate the Clipboard into an object that will solve as much of the problem as possible in a reusable class.


Having the Clipboard support multiple formats is an important consideration. Even if most of the time you intend to supply just cf_Text items to the Clipboard, if you ever want to include a bitmap as well, but the class can’t support it, it won’t have done you much good. That’s why it’s important to support multiple formats at the outset. 


On the other hand, the implementation should not pose any extra burden when text is all you need to support.


Another aspect of multiple formats we’ll have to contend with is that of format negotiation, where the Clipboard source and recipient agree in which format the item should be rendered. If your application wants to get an item from the Clipboard, and the item exists in five different formats, which one should your application take? If your application only supports one format, the answer is obvious, but suppose it supports all five. You’ll want to take the format with the least loss of data. When you design your program, you’ll know which that will be, but we can’t guess now. The ability to specify format priority should be built into our tClipboard class.


It turns out that each format can best be managed if we create a format class and a tClipboard class that is a collection of format class objects. For a simple application, the only format placed in the collections might be cf_Text, but your app could place as many formats there as needed, in priority order. Each format object should include methods for rendering (placing into the Clipboard) and garnering (gathering from the Clipboard) data in the appropriate format.


How can we code an object that will place something into the Clipboard, when we don’t know what that something will be? This is similar to the problem ObjectWindow’s designers faced when writing tDlgWindow. The designers needed to know what Windows class name to give the physical dialog box, but, at the time they were writing the class, they couldn’t possibly know. So they wrote a virtual method, GetClassName, which reaches out and takes the information. You are expected to override this method with one that will return the required class name; since it is a virtual method, your overriding method will be invoked instead of the abstract place holder.


We can use a similar technique for our format object class.


The tClipboard class itself becomes a derivative of tCollection, with enough added functionality to manage the broader aspects of the Clipboard, leaving the tClipboardFormat derivatives to do the actual dirty work.


Creating an Object for Each Format�tc "Creating an Object for Each Format"�


As a collection of format objects, the tClipboard class will manage the broader aspects of the Clipboard while letting the format objects render and garner the actual data. Virtual methods ensure that your application, using derived tClipboardFormat objects, will be able to place and retrieve the data it needs.


Copy CLASSKEL.PAS as CLPBOARD.PAS and make the necessary changes to create the tClipboardFormat definition. You can remove the Load and Store methods and calls to RegisterType; neither the tClipboardFormat nor tClipboard objects will be streamable. At this point, the Interface section of CLPBOARD.PAS should look like this:





Unit Clpboard;





	(***********************************************)


						Interface


	(***********************************************)





	Uses


		Objects,


		OWindows,


		Win31,


		WinTypes,


		WinProcs,


		Strings,


		OString;





	Type


		pClipboardFormat = ^tClipboardFormat;


		tClipboardFormat = Object (tObject)


			Format: Word;


			Constructor Init;


			Constructor InitStandard (aFormat: Word);


			Constructor InitCustom (aName: pChar);


			Procedure Render


				(


				Parent: pWindowsObject;


				var Buffer: Pointer;


				var BufferLength: LongInt


				); Virtual;


			Procedure Garner


				(


				Parent: pWindowsObject;


				Buffer: Pointer;


				BufferLength: LongInt


				); Virtual;


			Function RenderHandle


				(


				Parent: pWindowsObject


				): tHandle; Virtual;


			Procedure GarnerHandle


				(


				Parent: pWindowsObject;


				h: tHandle


				); Virtual;


			End;





The one property, Format, is not an object, so we don’t need an explicit destructor. In normal use, you will need to specify one of the standard formats; for this we have an InitStandard constructor:





Constructor tClipboardFormat.InitStandard (aFormat: Word);


	Begin


	Inherited Init;


	Format := aFormat;


	End;





The aFormat parameter requires one of the pre-defined constants in WinTypes, such as cf_Text or cf_Bitmap. These are documented in BPW’s online help under Clipboard Formats.


Since tClipboardFormat is derived from tObject (so that it can be made part of a tCollection), it will inherit an Init constructor, unless we supply one. Since the most common format is undoubtedly cf_Text, that format makes a good default:





Constructor tClipboardFormat.Init;


	Begin


	Inherited Init;


	Format := cf_Text;


	End;





One of the strengths of the Windows Clipboard is that the number of formats it supports is not fixed. You can define your own formats, if you wish. A common use of a private format is to encode data for efficient pasting back into your own application. A name and address application, for instance, could place a copy of the entire internal data structure. This structure wouldn’t have any meaning to another application, but much cutting and pasting takes place within a single application anyway.


To create a tClipboardFormat object for a custom format, use the InitCustom constructor:





Constructor tClipboardFormat.InitCustom (aName: pChar);


	Begin


	Inherited Init;


	Format := RegisterClipboardFormat (aName);


	End;





The Render virtual method delivers data to the Clipboard in a specific format. The default method, which is intended to be overridden, cannot put anything meaningful into the Clipboard. Still, it’s considered good form to supply abstract methods that, if they can do no good, at least do no harm. So the default Render method aims the Buffer pointer at an empty string, and sets the length to zero:





Procedure tClipboardFormat.Render


		(


		Parent: pWindowsObject;


		var Buffer: Pointer;


		var BufferLength: LongInt


		);


	Begin


	Buffer := pChar('');


	BufferLength := 0;


	End;





The Parent pointer provides a backlink to your application. Although �it is defined as a pWindowsObject, you can always cast it to a pointer to your derived tDlgAppWindow to access properties and methods.


Note that this method does not supply you with a valid pointer. Presumably your application already has the data to be copied; just put the address of that data into Buffer and set BufferLength; tClipboardFormat will do the rest.


Another caveat: When you override Render, do not give Buffer the address of a local variable! When Render returns to its caller, that local variable will no longer exist, and its address will be meaningless. In the abstract method above, Buffer is made to point at a byte in the constants segment.


The calling sequence of Render is actually a compromise, because the Clipboard itself does not deal with data buffers. It only understands global data handles, which are an entirely different thing. For most formats, dealing with buffers and arrays of data bytes will be most natural; but there are several formats that require you to use the actual handle. These include cf_Bitmap, cf_DIB (device independent bitmap), and cf_Palette. Therefore, there needs to be another method that provides the interface between handles and data buffers. That method is RenderHandle:





Function tClipboardFormat.RenderHandle


		(


		Parent: pWindowsObject


		): tHandle;


	Var


		h: tHandle;


		Buffer: Pointer;


		Original: Pointer;


		OriginalLength: LongInt;


	Begin


	Render (Parent, Original, OriginalLength);


	h := GlobalAlloc (gmem_DdeShare, OriginalLength);


	Buffer := GlobalLock (h);


	Move (Original^, Buffer^, OriginalLength);


	GlobalUnlock (h);


	RenderHandle := h;


	End;





As you can see, this method invokes Render. In most cases, it will be Render that you override, not RenderHandle. But for cf_Bitmap format objects and the others mentioned, you will override RenderHandle instead.


Render gives us a pointer to the data to be sent to the Clipboard, and a count of bytes to send. We use that count as a parameter to GlobalAlloc, which provides us with the handle to the memory block. The gmem_DdeShare tells GlobalAlloc to give us a block that may be shared among applications. Strickly speaking, this isn’t required because when Clipboard takes over ownership of the block it will change the block attributes anyway. But we’ll be using the tClipboardFormat class for another purpose in Chapter 11, and the gmem_DdeShare flag will be required then.


Understanding Global Handles


We have dealt with handles in a relaxed sort of way until now; but with the appearance of GlobalAlloc and its cousins, we’ll now go into more detail. 


I mentioned in Chapter 3 that there are three kinds of heaps used by BPW: local, global, and the Borland Pascal heap that sub-allocates blocks of global heap.


When Windows was first designed, the only PCs available had 8088 or 8086 CPUs. These ran in what we now call “real mode” and were limited to 1Mb of memory.


It was clear from the start that Windows and its applications would require much more memory than that, so the designers implemented an overlay scheme involving the segments compiled programs were already divided into. Unless otherwise marked, segments were only loaded into memory when needed, and could be discarded if the memory they occupied was needed for something else. This implied that segments might also be moved in memory, which meant traditional pointers would not provide a reliable means of accessing them. Instead, Windows relied on handles, 16-bit values that were actually indexes into tables where a description of the memory block was kept. There were several different tables, each used to store a different kind of object. Thus, we have window handles, handles to bitmaps and brushes, instance handles, module handles, and so on. Among this plethora of handle types are memory handles: global and local.


Global handles are created using GlobalAlloc; a pointer to the allocated memory is obtained from GlobalLock and released by GlobalUnlock. Finally, when the memory block is no longer needed, the handle is passed to GlobalFree to be released.


There is an identical set of routines with the prefix Local instead of Global for managing the local heap.


It is absolutely essential that these functions be called in pairs. Every block that is allocated must eventually be freed; every pointer that is locked must be unlocked. You are not supposed to leave a pointer locked between operations; so programmers did not usually save the pointers, just the handles. As each operation started, the handle was used to lock a pointer, the data accessed, and the handle unlocked.


GlobalAlloc and the other global heap management functions are still used, but since Windows 3.1 cannot run in real mode, there is no longer any harm in leaving pointers locked. (Windows can move memory blocks around in protected mode without changing the segment part of the address.) That’s what makes BPW’s GetMem work: The Borland Pascal memory sub-allocation scheme has to use GlobalAlloc to get the global block to sub-allocate. But it does the tricky part of managing locks and handles for you.








We pass the handle returned by GlobalAlloc to GlobalLock, then assign the address returned by GlobalLock to Buffer, then invoke the standard Move procedure to copy the original data to the new memory block. Finally, we unlock the handle and return it to the caller.


Normally the caller of GlobalAlloc must eventually call GlobalFree. However, in Clipboard use, when we give the handle to the Clipboard we also pass on ownership of the block. The Clipboard itself will free that memory block when it is no longer needed.


For getting data from the Clipboard, you will usually override the Garner method. The abstract method doesn’t do anything:





Procedure tClipboardFormat.Garner


	(


	Parent: pWindowsObject;


	Buffer: Pointer;


	BufferLength: LongInt


	);


	Begin


	End;





As Render is called from RenderHandle, Garner is called from GarnerHandle:





Procedure tClipboardFormat.GarnerHandle


		(


		Parent: pWindowsObject;


		h: tHandle


		);


	Var


		Buffer: Pointer;


	Begin


	Buffer := GlobalLock (h);


	Garner (Parent, Buffer, GlobalSize (h));


	GlobalUnlock (h);


	End;





In this case, the handle is supplied by the Clipboard; we only have to lock it to get a pointer and invoke the overriding Garner routine, which presumably will do something with it. We then unlock the handle, but we don’t free it—it’s not ours to free; the Clipboard owns it now.


There are times when an application would like to know a given format’s name. A Clipboard viewer certainly needs this information; it can use it to label the displayed data. Some applications enumerate all available formats and give the user the choice of which to paste.�tc "There are times when an application would like to know a given format’s name. A Clipboard viewer certainly needs this information; it can use it to label the displayed data. Some applications enumerate all available formats and give the user the choice of which to paste."�


The names of custom formats are easy to retrieve; you simply invoke GetClipboardFormatName. But we are told we must not call this function with an identifier for one of the standard formats. Thus, in the GetFormatName method, we check for each of the standard formats before resorting to GetClipboardFormatName:





Procedure tClipboardFormat.GetFormatName (Var FormatName: tOString);


	Var


		Buffer: Array [0..64] of Char;


	Begin


	Case Format of


		cf_Text:


			FormatName.SetTextP ('cf_Text');


		cf_Bitmap:


			FormatName.SetTextP ('cf_Bitmap');


		cf_MetaFilePict:


			FormatName.SetTextP ('cf_MetaFilePict');


		cf_SYLK:


			FormatName.SetTextP ('cf_SYLK');


		cf_DIF:


			FormatName.SetTextP ('cf_DIF');


		cf_TIFF:


			FormatName.SetTextP ('cf_TIFF');


		cf_OEMText:


			FormatName.SetTextP ('cf_OEMText');


		cf_DIB:


			FormatName.SetTextP ('cf_DIB');


		cf_Palette:


			FormatName.SetTextP ('cf_Palette');


		cf_PenData:


			FormatName.SetTextP ('cf_PenData');


		cf_RIFF:


			FormatName.SetTextP ('cf_RIFF');


		cf_Wave:


			FormatName.SetTextP ('cf_Wave');


		cf_OwnerDisplay:


			FormatName.SetTextP ('cf_OwnerDisplay');


		cf_DSPText:


			FormatName.SetTextP ('cf_DSPText');


		cf_DSPBitmap:


			FormatName.SetTextP ('cf_DSPBitmap');


		cf_DSPMetaFilePict:


			FormatName.SetTextP ('cf_DSPMetaFilePict');


		else


			Begin


			GetClipboardFormatName (Format, Buffer, SizeOf (Buffer));


			FormatName.SetTextC (Buffer);


			End;


		End;





	End;





One of the peculiarities of GlobalAlloc is that it usually gives you a little more memory than you asked for. GlobalSize usually does not equal your allocation request exactly (although it will never be smaller). Therefore, you should not take the BufferLength parameter to Garner too seriously. If, for example, you are copying data into a structure, you should use the SizeOf function to determine how many bytes to copy rather than BufferLength.


This is a constant problem with cf_Text format, so we are going to derive class tClipboardText. Although this is also an abstract class, it will help solve the block size problem for NULL-terminated strings. Add the following definition:





Type


	pClipboardText = ^tClipboardText;


	tClipboardText = Object (tClipboardFormat)


		Procedure GarnerHandle


			(


			Parent: pWindowsObject;


			h: tHandle


			); Virtual;


		End;





and the following method:





Procedure tClipboardText.GarnerHandle


		(


		Parent: pWindowsObject;


		h: tHandle


		);


	Var


		Buffer: Pointer;


	Begin


	Buffer := GlobalLock (h);


	Garner (Parent, Buffer, StrLen (Buffer));


	GlobalUnlock (h);


	End;





The only difference is that the StrLen function from the Strings unit is used to supply Garner with the BufferLength.


Let me emphasize: These are abstract methods. You will have to derive at least one new class from tClipboardFormat or tClipboardText if you want to use the tClipboard class. We’ll do that shortly, when we write the Word Counter application.


Supporting Bitmap and Metafile Formats�tc "Supporting Bitmap and Metafile Formats"�


Next to text, the most common formats to be placed on the Clipboard are bitmap and metafile. In this section we’ll create abstract classes to support these formats.


As a graphical user interface, Windows primarily supports two formats for representing images on-screen. A bitmap is an array of values which describe each pixel in the image. A metafile, in contrast, is a block of instructions for drawing the image.


Once an appropriate device context has been created, the same drawing instructions that will produce a bitmap can also produce a metafile. Therefore, the task of the two new formats will be to create those device contexts.


The definitions of the two formats are as follows. Note that tClipboardMetafile is derived from tClipboardBitmap:





Type


	pClipboardBitmap = ^tClipboardBitmap;


	tClipboardBitmap = Object (tClipboardFormat)


		Bounds: Record


			Width, Height: Integer;


			End;


		Constructor Init (aWidth, aHeight: Integer);


		Function RenderHandle


			(Parent: pWindowsObject): tHandle; Virtual;


		Procedure RenderDrawing


			(Parent: pWindowsObject; DC: hDC); Virtual;


		End;





Type


	pClipboardMetafile = ^tClipboardMetafile;


	tClipboardMetafile = Object (tClipboardBitmap)


		Constructor Init (aWidth, aHeight: Integer);


		Function RenderHandle


			(Parent: pWindowsObject): tHandle; Virtual;


		End;





The constructor for tClipboardBitmap must initialize the only property, a Bounds structure that indicates the size of the bitmap:





Constructor tClipboardBitmap.Init (aWidth, aHeight: Integer);


	Begin


	Inherited InitStandard (cf_Bitmap);


	Bounds.Width := aWidth;


	Bounds.Height := aHeight;


	End;





RenderHandle is the next method, but before we can look at it we need to inspect an unpublished method, SizeToHiMetric:





Procedure SizeToHiMetric (var Width, Height: Integer);


	Const


		HiMetricPerInch : Longint = 2540;


	Var


		DC: HDC;


		DpiX, DpiY: Integer;


	Begin


	DC := GetDC(0);


	DpiX := GetDeviceCaps (DC, LogPixelsX);


	DpiY := GetDeviceCaps (DC, LogPixelsY);


	Width := Round (Width  * HiMetricPerInch / DpiX);


	Height:= Round (Height * HiMetricPerInch / DpiY);


	ReleaseDC (0, DC);


	End;





Bitmaps include a dimension property which must contain the size of the bitmap in mm_HiMetric map mode units. SizeToHiMetric performs the needed calculations.


With that out of the way, we can study RenderHandle:





Function tClipboardBitmap.RenderHandle


		(


		Parent: pWindowsObject


		): tHandle;


	Var


		DC: hDC;


		Bitmap,


		OldBitmap: hBitmap;


		Width, Height: Integer;


	Begin


	DC := GetDC (Parent^.hWindow);


	Width := Bounds.Width;


	Height := Bounds.Height;


	Bitmap := CreateCompatibleBitmap (DC, Width, Height);


	ReleaseDC (Parent^.hWindow, DC);





	DC := CreateCompatibleDC (0);


	OldBitmap := SelectObject (DC, Bitmap);


	PatBlt (DC, 0, 0, Width, Height, Whiteness);


	SetWindowOrg (DC, 0, 0);


	SetWindowExt (DC, Width, Height);


	RenderDrawing (Parent, DC);


	SelectObject (DC, OldBitmap);


	DeleteDC (DC);





	SizeToHiMetric (Width, Height);


	SetBitmapDimension (Bitmap,


		Round (Width / 10), Round (Height / 10));


	RenderHandle := Bitmap;


	End;





The code is organized into three blocks. In the first, a “compatible bitmap” is created. The bitmap is compatible, that is, it has the same characteristics, as the display screen. In the second block, a compatible device context is also created. This device context is not attached to any device; but when we select the bitmap into it, any drawing we do to this device context will be painted on the selected bitmap. So we paint the bitmap white (it was originally “undefined”) and invoke RenderDrawing so the derived class can actually create the desired images. Finally, the bitmap dimensions are calculated and set, and the bitmap handle is returned.


The abstract RenderDrawing method does nothing:





Procedure tClipboardBitmap.RenderDrawing


		(Parent: pWindowsObject; DC: hDC);


	Begin


	End;





The tClipboardMetafile format is so similar, it could be derived from tClipboardBitmap. We do replace the constructor so it can report the proper format type:





Constructor tClipboardMetafile.Init (aWidth, aHeight: Integer);


	Begin


	Inherited InitStandard (cf_MetafilePict);


	Bounds.Width := aWidth;


	Bounds.Height := aHeight;


	End;





Function tClipboardMetafile.RenderHandle


		(


		Parent: pWindowsObject


		): tHandle;


	Var


		Picture: pMetaFilePict;


		h,


		Metafile: tHandle;


		DC: hDC;


		Width, Height: Integer;


	Begin


	DC   := CreateMetaFile (nil);


	Width := Bounds.Width;


	Height := Bounds.Height;


	SetWindowOrg (DC, 0, 0);


	SetWindowExt (DC, Width, Height);


	RenderDrawing (Parent, DC);


	Metafile := CloseMetaFile (DC);





	h := GlobalAlloc (gmem_DdeShare, SizeOf (tMetaFilePict));


	If h <> 0 then


		Begin


		SizeToHiMetric (Width, Height);


		Picture := pMetaFilePict (GlobalLock(h));


		Picture^.mm := mm_Anisotropic;


		Picture^.hMF := Metafile;


		Picture^.xExt := Width;


		Picture^.yExt := Height;


		GlobalUnlock (h);


		End;


	RenderHandle := h;


	End;





For metafiles, the RenderHandle method is broken into two blocks. In the first, a metafile device context is created and passed to RenderDrawing. To RenderDrawing there will seem to be no difference—a DC is a DC. In the second block, a metafile header is created; a handle to the real metafile is copied to the header and it is the header handle that is returned.


Maintaining a Collection of Formats�tc "Maintaining a Collection of Formats"�


When a program supplies data to the Clipboard, it may be supplied in a variety of formats. The receiving program is supposed to ask for the most sophisticated—the documentation calls it the most “complete”—format first. If that is not available, it asks for the next most complete, and so on. But who’s to say which is most complete? Is a picture (metafile) of “fancy text” more complete than Rich Text Format? By implementing tClipboard as a collection of tClipboardFormat objects, we get format negotiation as a free bonus.


The tClipboard class is defined as:





Type


	pClipboard = ^tClipboard;


	tClipboard = Object (tCollection)


		Parent: pWindowsObject;


		Constructor Init (aParent: pWindowsObject);


		Procedure SetData;


		Procedure GetData;


		Function IsData: Boolean;


		End;





If that seems unbelievably simple to you, remember that much of the work has already been done—in the tClipboardFormat class—and much is done by tClipboard’s ancestor, tCollection.


The Parent property will provide the backlink to the parent window in the tClipboardFormat objects’ Garner and Render methods. It is set in the constructor, but since it is just a pointer to an object tClipboard does not own, there is no destructor:





Constructor tClipboard.Init (aParent: pWindowsObject);


	Begin


	Inherited Init (1, 5);


	Parent := aParent;


	End;





The Init constructor, inherited from tCollection, wants to know how many elements to make room for to start, and how many elements it should add space for each time the current number is inadequate. Most applications only use cf_Text format, so I made the starting value 1.


The SetData method will be called to place data into the Clipboard:





Procedure tClipboard.SetData;


	Procedure SetFormattedData (Format: pClipboardFormat); Far;


		Begin


		SetClipboardData


			(Format^.Format, Format^.RenderHandle (Parent));


		End;


	Begin


	If OpenClipboard (Parent^.hWindow) then


		Begin


		EmptyClipboard;


		ForEach (@SetFormattedData);


		CloseClipboard;


		End;


	End;





When writing to the Clipboard, the first step is to open it. OpenClipboard can only fail if another window currently has the Clipboard open. That means it can’t fail (barring program bugs) in Windows 3.1; but the introduction of Windows NT is right around the corner, which will allow more than one application to process messages concurrently. Therefore, to be on the safe side, check the return from OpenClipboard.


If you place two items in the Clipboard in the same format, the second will replace the first. EmptyClipboard covers the situation where your application does not replace all formats. If you neglect to invoke EmptyClipboard, you can leave your users in the strange position of having, say, text from your application in the Clipboard but being unable to paste it into a drawing program because there’s a bitmap there that’s left over from a previous Edit..Copy. Don’t laugh; there’s one major drawing package in release that has this very problem.


By now, you should be familiar with the ForEach method that allows the tClipboard object to enumerate the formats, invoking each one’s RenderHandle method in turn. Finally, after all formats are heard from, access to the Clipboard is closed.


When you supply data in more than one format, most other applications will assume you have sent the formats in order of increasing data loss. You will have to decide what that means. If your application is primarily text-oriented but you supply a bitmapped picture of your text as a courtesy, your cf_Text format should be the first posted. On the other hand, if you post a bitmap and, for text, the name of the bitmap, clearly cf_Bitmap is more complete.


Whatever you decide, simply add tClipboardFormat objects to your tClipboard collection in priority order (using the tCollection.Insert method), and the ForEach method will ensure your data is placed in the correct sequence.


To get data from the Clipboard, invoke the GetData method:





Procedure tClipboard.GetData;


	Function GetFormattedData


		(Format: pClipboardFormat): Boolean; Far;


	Begin


		If IsClipboardFormatAvailable (Format^.Format) then


			Begin


			Format^.GarnerHandle (Parent,


				GetClipboardData (Format^.Format));


			GetFormattedData := True;


			End


		else


			GetFormattedData := False;


		End;


	Begin


		If OpenClipboard (Parent^.hWindow) then


		Begin


		FirstThat (@GetFormattedData);


		CloseClipboard;


		End;


	End;





Instead of ForEach, we enumerate the formats using FirstThat because we only want to retrieve the data once, in the preferred format. Again, formats will be checked in the order in which the tClipboardFormat objects were inserted; and once a format has been accepted, the Clipboard can be closed.


The next tClipboard method will be invoked by a wmMenuPopup handler to determine whether the Edit..Paste command should be enabled. It simply determines whether any suitably formatted data can be found in the Clipboard:





Function tClipboard.IsData: Boolean;


	Function HasFormattedData


			(Format: pClipboardFormat): Boolean; Far;


		Begin


		If IsClipboardFormatAvailable (Format^.Format) then


			Begin


			HasFormattedData := True;


			End


		else


			HasFormattedData := False;


		End;


	Var


		Format: pClipboardFormat;


	Begin


	Format := FirstThat (@HasFormattedData);


	IsData := Assigned (Format);


	End;





This method is very similar to GetData, except that it doesn’t actually get any data—just checks to see if any is there.


Finally, it is occasionally desirable for an application to be able to clear the Clipboard without replacing its contents with anything else. The Clear method accomplishes this:





Procedure tClipboard.Clear;


	Begin


	If OpenClipboard (Parent^.hWindow) then


		Begin


		EmptyClipboard;


		CloseClipboard;


		End;


	End;


Using a tClipboard Object�tc "Using a tClipboard Object"�


Now that the tClipboard class has been defined, we can explore ways of using it. Although the class is not tied into tDlgAppWindow, it can be used there to provide Clipboard access to a tDlgAppWindow-derived �application.


Adding a tClipboard Object to tDlgAppWindow�tc "Adding a tClipboard Object to tDlgAppWindow"�


Although some applications may not use the Clipboard, the overhead of including a tClipboard property is so small there’s no reason not to do it.


To start with, we’ll have to add ClpBoard to the list of units in the Uses clause:





Uses


			 


			 


	ClpBoard;





In addition to two enhancements to several existing methods, we must also add two new properties and five new methods to the tDlgAppWindow definition:





Type


	pDlgAppWindow = ^tDlgAppWindow;


	tDlgAppWindow = Object (tDlgWindow)


			 


			 


		Clipboard: tClipboard;


		EditPasteMenu: pMenuItem;


			 


			 


		Function GetEditPasteMenuID: Word; Virtual;


		Procedure cmEditCopy (var Msg: tMessage);


			Virtual cm_First + cm_EditCopy;


		Procedure cmEditCut (var Msg: tMessage);


			Virtual cm_First + cm_EditCut;


		Procedure cmEditPaste (var Msg: tMessage);


			Virtual cm_First + cm_EditPaste;


		Procedure cmEditClear (var Msg: tMessage);


			Virtual cm_First + cm_EditClear;


		End;





The Clipboard property is obvious enough. The EditPasteMenu property is similar to the FileSaveMenu property; it represents a menu item that should be enabled or disabled, depending on the presence of data in the Clipboard. Both Clipboard and EditPasteMenu must be initialized:





Constructor tDlgAppWindow.Init;


	Begin


			 


			 


	Clipboard.Init (@Self);


	EditPasteMenu := Nil;


	End;





And like the File..Save command, not all applications may have an Edit..Save command. An application that does not must override the GetEditPasteMenuID method to return zero. The default method returns the standard constant:





Function tDlgAppWindow.GetEditPasteMenuID: Word;


	Begin


	GetEditPasteMenuID := cm_EditPaste;


	End;





The SetupWindow method must be enhanced to initialize EditPasteMenu:





Procedure tDlgAppWindow.SetupWindow;


	Var


		Menu: tHandle;


	Begin


			 


			 


	If GetEditPasteMenuID > 0 then


		Begin


		Menu := GetSubMenu (GetMenu (Self.hWindow), 1);


		EditPasteMenu :=


			New (pMenuItem, InitResource (Menu, GetEditPasteMenuID));


		End;


	End;





The new code parallels exactly the code for the FileSaveMenu property. Likewise, the new properties must be destroyed:





Destructor tDlgAppWindow.Done;


	Begin


			 


			 


	Clipboard.Done;


	If Assigned (EditPasteMenu) then


		Dispose (EditPasteMenu, Done);


	Inherited Done;


	End;





The wmInitMenuPopup method must now check for the new menu object:





Procedure tDlgAppWindow.wmInitMenuPopup (var Msg: tMessage);


	Begin


			 


			 


	If Assigned (EditPasteMenu) then


		If Msg.wParam = EditPasteMenu^.ParentMenu then


			If Clipboard.IsData then


				EditPasteMenu^.Enable


			else


				EditPasteMenu^.Disable;


	End;





The remaining new methods are all command handlers:





Procedure tDlgAppWindow.cmEditCopy (var Msg: tMessage);


	Begin


	Clipboard.SetData;


	End;





Procedure tDlgAppWindow.cmEditCut (var Msg: tMessage);


	Begin


	cmEditCopy (Msg);


	cmEditClear (Msg);


	End;





Procedure tDlgAppWindow.cmEditPaste (var Msg: tMessage);


	Begin


	Clipboard.GetData;


	End;





Procedure tDlgAppWindow.cmEditClear (var Msg: tMessage);


	Begin


	End;





cmEditCut simply passes the buck to cmEditCopy and cmEditClear, because, in terms of the Clipboard, cutting and copying are identical. You’ll have to add whatever code is needed to delete a cut item when you override the abstract cmEditClear method.


Creating a tClipboard Application�tc "Creating a tClipboard Application"�


Now we get to see the tClipboard class in action. The WordCnt program accepts a block of text from the Clipboard and counts the words in it, while displaying a running count on screen. When it’s done, the count can be pasted back into the Clipboard.


First, create WORDCNT.RES from DLGSKEL.RES. The About box should be modified appropriately; the MAIN dialog should be made to look like the one in Figure 9.1.

Figure 9.1: Word Counter main dialog design.


The only accelerators you’ll want to keep are Ctrl+Ins, Shift+Ins, and F1.


In Edit as Text format, the menu should be defined as:





MAIN MENU PRELOAD MOVEABLE DISCARDABLE 


BEGIN


	POPUP "&File"


	BEGIN


		MENUITEM "&Open...", 102


		MENUITEM SEPARATOR


		MENUITEM "E&xit", 24340


	END





	POPUP "&Edit"


	BEGIN


		MENUITEM "&Paste and Count\tShift+Ins", 203


		MENUITEM "&Copy Count\tCtrl+Ins", 202


	END





	POPUP "&Help"


	BEGIN


		MENUITEM "&Index\tF1", 901


		MENUITEM SEPARATOR


		MENUITEM "&About Word Counter...", 999


	END


END





In Stringtable 600, string 600 should be changed to read, “Text Files |*.txt|Formatted text files|*.asc|All Files(*.*)|*.*|” and string 601 should be “txt”. In addition, several new strings should be added, starting with string ID 1:





STRINGTABLE LOADONCALL MOVEABLE DISCARDABLE 


BEGIN


	1, "Receiving Clipboard data..."


	2, "Loading file data..."


	3, "Counting words..."


	4, "Done."


	5, "<Clipboard>"


	6, "Sorry, there's not enough memory to satisfy your request. Close other applications and try again."


END





Finally, I designed the icon shown in Figure 9.2.

Figure 9.2. The WORDCNT Icon.


Copy DLGSKEL.PAS to create WORDCNT.PAS. The file should open thus:





Program WordCnt;





	{$R WordCnt.res}





	Uses


		Objects,


		OWindows,


		WinProcs,


		WinTypes,


		DlgApp,


		Controls,


		Clpboard,


		OString,


		Strings;





	Const


		id_Display = 1001;





	Const


		str_Rendering = 1;


		str_Loading = 2;


		str_Counting = 3;


		str_Done = 4;


		str_Clipboard = 5;


		str_OutOfMemory = 6;





Remembering that we must always derive at least one class from tClipboardFormat or tClipboardText if we want to use the Clipboard, we can define tClipText:





Type


	pClipText = ^tClipText;


	tClipText = Object (tClipboardFormat)


		Procedure GarnerHandle


			(Parent: pWindowsObject; h: tHandle); Virtual;


		End;





 Even though it is the cf_Text format we’ll be interested in, we are not deriving this class from tClipboardText. The reason why will become apparent shortly. But first, let’s continue with the tMainDlg and tDlgApp class definitions:





Type


	pMainDlg = ^tMainDlg;


	tMainDlg = Object (tDlgAppWindow)


	Display: tXStatic;


	Count: LongInt;


	CountText: tOString;


	Constructor Init;


	Destructor Done; Virtual;


	Function GetFileSaveMenuID: Word; Virtual;


	Procedure cmEditPaste (var Msg: tMessage);


		Virtual cm_First + cm_EditPaste;


	Procedure CountWords (h: tHandle);


	End;





Type


	pDlgApp = ^tDlgApp;


	tDlgApp = Object (tApplication)


	Procedure InitMainWindow; Virtual;


	End;





Var


	MyDlgApp: tDlgApp;





Our first look at using the Clipboard comes with the Init constructor:





Constructor tMainDlg.Init;


	Begin


	Inherited Init;


	Clipboard.Insert (New (pClipText, Init));


	Display.InitResource (@Self, id_Display, 16);


	CountText.Init (8);


	CountText.SetTextN (0);


	End;





We allocate one instance of tClipText and insert its pointer into the Clipboard collection that tMainDlg inherited from tDlgAppWindow. We do not have to explicitly dispose of the format because Done, inherited from tDlgAppWindow, will dispose of Clipboard, and that will automatically dispose of any elements inserted into it. Our Done just deals with tMainDlg’s properties:





Destructor tMainDlg.Done;


	Begin


	Display.Done;


	CountText.Done;


	Inherited Done;


	End;





Since Word Counter cannot save a file, there is no File..Save menu item. We must therefore override GetFileSaveMenuID:





Function tMainDlg.GetFileSaveMenuID: Word;


	Begin


	GetFileSaveMenuID := 0;


	End;





We must also override the RegisterApp method. Although we want Word Counter, as an application, to appear in the Registration Database, we don’t want Windows to automatically start Word Counter when the user double-clicks on the name of a .txt file. So we override tDlgAppWindow.RegisterApp with a copy of the inherited method that does not change the entry for .txt files:





Procedure tMainDlg.RegisterApp;


	Var


		ClassKey,


		CommandKey: hKey;


		ClassName: tOString;


	Procedure OpenKeys;


		Begin


		RegCreateKey (HKEY_CLASSES_ROOT,


			ClassName.CString, ClassKey);


		RegCreateKey (ClassKey, 'shell\open\command', CommandKey);


		End;


	Procedure SetValues;


		Var


			Text: tOString;


		Begin


		RegCheckValue (ClassKey, BaseCaption);


		Text.Init (128);


		GetCommandFormat (Text);


		RegCheckValue (CommandKey, Text);


		Text.Done;


		End;


	Procedure CloseKeys;


		Begin


		RegCloseKey (ClassKey);


		RegCloseKey (CommandKey);


		End;


	Begin


	ClassName.InitTextC (Application^.Name);


	OpenKeys;


	SetValues;


	CloseKeys;


	ClassName.Done;


	End;





We’ve also taken out the call to RegisterAppExtended, since tMainDlg won’t be further subclassed and we know tMainDlg doesn’t need it. (In order to compile this copied code, don’t forget to add ShellAPI to the list of units in the Uses clause.)


Finally, we get to tMainDlg’s handler for the Edit..Paste command:





Procedure tMainDlg.cmEditPaste (var Msg: tMessage);


	Begin


	CountText.SetTextN (0);


	Display.SetText (CountText);


	DataFile.Title.SetTextR (str_Clipboard);


	SetCaption;


	Help.SetTextR (str_Garnering);


	Inherited cmEditPaste (Msg);


	End;





When this command is executed, we set the displayed count to zero, fool the DataFile property inherited from tDlgAppWindow into thinking the current filename is “<Clipboard>” so we can use that in setting the caption, cause the status bar to announce we are receiving text from the Clipboard, and invoke the ancestor cmEditPaste method. That method, you’ll recall, simply invokes Clipboard.GetData, which will enumerate the formats—we’ve only defined the one—and invoke each one’s GarnerHandle method.


Next, we’ll switch classes and look at tClipText.GarnerHandle:





Procedure tClipText.GarnerHandle


		(Parent: pWindowsObject; h: tHandle);


	Begin


	pMainDlg(Parent)^.CountWords (h);


	End;





Kind of like table tennis, isn’t it? But this kind of back-and-forth is not unusual in an object-oriented program. All this method does is invoke the main window’s CountWords method:





Procedure tMainDlg.CountWords (h: tHandle);


	Var


		P: pChar;


		InAWord: Boolean;


	Procedure DisplayCount;


		Begin


		CountText.SetTextN (Count);


		Display.SetText (CountText);


		End;


	Begin


	Help.SetTextR (str_Counting);


	P := GlobalLock (h);


	InAWord := False;


	Count := 0;


	While P^ <> #0 do


		Begin


		If P^ in [ 'A'..'Z', 'a'..'z', '''', '0'..'9', '-'] then


			Begin


			If not InAWord then


				Begin


				Inc (Count);


				If (Count mod 1000) = 0 then


					DisplayCount;


				InAWord := True;


				End;


			End


		else


			InAWord := False;


		P := @P[1];


		End;


	DisplayCount;


	GlobalUnlock (h);


	Help.SetTextR (str_Done);


	End;





When the method starts, the status bar announces that the counting has begun; when it’s over, the status bar says so. We only update the display once every 1000 words; much more often than that and the counting operation is slowed too much by the display.


You may have noticed we made the status bar read “Getting data from Clipboard...”, then “Counting words...”, with almost no time elapsing between the two. Surely the user won’t have time to read the first message!


Well, that’s usually true. However, an application is not required to actually place data in the Clipboard during execution of RenderHandle. If the “handle” has a value of zero, the Clipboard interprets that to mean the application will deliver the data later, if asked. This is called delayed rendering. It is done if rendering the data for the Clipboard in all supported formats would take an inordinate amount of time. Usually the application knows what it’s doing; so if delayed rendering is being practiced, the message “Getting data from Clipboard...” may well be visible long enough to read. When I place the manuscript to this book in the Clipboard and paste it in text format, it takes over a minute for Word for Windows to render it in that format.


The rest of the CountWords method is pretty straightforward, as word-counting algorithms go. There is just one question you are likely to have: Why do I increment the pointer instead of indexing it? I promise to explain that shortly.


At this point, given that the remainder of the WORDCNT.PAS file reads:





Procedure tDlgApp.InitMainWindow;


	Begin


	MainWindow := New (pMainDlg, Init);


	End;





Begin


MyDlgApp.Init ('WordCounter');


MyDlgApp.Run;


MyDlgApp.Done;


End.





you’ll find the program can be compiled and run. Copy some text into the Clipboard, choose the Edit..Paste and Count command, and a correct count should appear in the dialog.


To add the ability to copy the word count into the Clipboard, give tClipText the following method:





Procedure tClipText.Render


		(


		Parent: pWindowsObject;


		var Buffer: Pointer;


		var BufferLength: LongInt


		);


	Begin


	Buffer := pMainDlg(Parent)^.CountText.CString;


	BufferLength := pMainDlg(Parent)^.CountText.Length + 1;


	End;





Remember, I warned that the address given to the Buffer parameter must not be that of a stack (local) variable, and it isn’t. Instead, we give it the address of the buffer in the main dialog’s CountText property, which was initially set to zero and was updated each time the count changed in the display.


There are two things missing from this application to make it complete (not counting an online help file!). One is code to enable the File..Open command, which presumably will read text from a file and count it. We’ll handle that in the next section. The other is to deal with extended delayed rendering times in a more elegant manner.


The message we displayed was nice; at the least the user will have some idea what’s happening as the seconds tick by, no counting seems to take place, and the keyboard and mouse remain unresponsive. But really we should change the cursor to an hour glass like other applications do when a lengthy process is in effect.


As opposed to many resources available to each application, there is only one mouse indicator that all applications must share. Moreover, it may not be visible if the user does not have a mouse. To make the hour glass cursor appear, we must:





•	Obtain a handle to the hour glass cursor





•	Obtain a handle to the original cursor, whatever it was





•	Set the new cursor handle





•	Invoke ShowCursor, in case the cursor was invisible because the user doesn’t have a mouse





When the wait is over and it’s time to change back to the old cursor, we basically follow the same steps in reverse:





•	Invoke ShowCursor with a False parameter, to remove the cursor if the user doesn’t have a mouse





•	Restore the original cursor handle





It’s not that many steps and a non-object-oriented program might repeat this code in a dozen places; but we’d like to do better than that. We can certainly create a tWaitCursor object; the only trick will lie in making it smart enough so it can be called by nested methods, each of which thinks a wait cursor would be appropriate. This can be accomplished with a simple Count property.


Copy CLASSKEL.PAS as WAIT.PAS:





Unit Wait;





	(***********************************************)


						Interface


	(***********************************************)





	Uses


		WinTypes,


		WinProcs,


		Objects;





	Type


		pWaitCursor = ^tWaitCursor;


		tWaitCursor = Object (tObject)


			Count: Byte;


			OldCursor: tHandle;


			Constructor Init;


			Procedure Start;


			Procedure Complete;


			End;





The Count property must be initialized; OldCursor doesn’t have to be. But neither property is an object so we don’t need a destructor:





Constructor tWaitCursor.Init;


	Begin


	Inherited Init;


	Count := 0;


	End;





The Start and Complete methods are mirror images of each other:





Procedure tWaitCursor.Start;


	Begin


	If Count = 0 then


		Begin


		OldCursor := SetCursor (LoadCursor (0, idc_Wait));


		ShowCursor (True);


		End;


	Inc (Count);


	End;





Procedure tWaitCursor.Complete;


	Begin


	Dec (Count);


	If Count = 0 then


		Begin


		ShowCursor (False);


		SetCursor (OldCursor);


		End;


	End;





As Start is called, perhaps many times by nested methods, it is only the first time that it loads the hour glass cursor and saves the old cursor handle—SetCursor and LoadCursor accomplish both tasks in a single statement. That is also the only time we invoke ShowCursor. ShowCursor has an internal counter; we could call it every time. But why bother, when any time after the first will have no other effect?


Complete undoes what Start did, restoring the old cursor and decrementing the ShowCursor counter if decrementing its own Count property resulted in bringing the count down to zero. Thus, as long as Start and Complete are always called in pairs, they can be nested as deeply as needed. When the outermost operation is Complete, the cursor will return to normal.


Now that we have a tWaitCursor class defined, we can add an object of that class to tDlgAppWindow:





Type


	pDlgAppWindow = ^tDlgAppWindow;


	tDlgAppWindow = Object (tDlgWindow)


			 


			 


		WaitCursor: tWaitCursor;


			 


			 


		End;





And, with that in place, we can add a nice, polite hour glass cursor to the paste operation in Word Counter:





Procedure tMainDlg.cmEditPaste (var Msg: tMessage);


	Begin


	WaitCursor.Start;


	CountText.SetTextN (0);


	Display.SetText (CountText);


	DataFile.Title.SetTextR (str_Clipboard);


	SetCaption;


	Help.SetTextR (str_Rendering);


	Inherited cmEditPaste (Msg);


	WaitCursor.Complete;


	End;





To see the hour glass cursor in operation, simply select a large block of text from any application and paste it into Word Counter. The wait cursor will provide the customary visual cue that the user must relax until the operation is complete.


Using Huge Pointers, or the End of the 64K Segment�tc "Using Huge Pointers, or the End of the 64K Segment"�


Although it is commonly believed that the architecture of the Intel CPUs used in IBM-compatible PCs prevents the use of memory segments greater than 64K, this isn’t true; it’s just more difficult. Windows makes huge segments available, and they can be accessed from Borland Pascal with just a little perseverance.


The Word Counter program works just fine until you try to place more than 64K of text into the Clipboard and then count the words in it. The program then enters an endless loop, counting and counting until a range error (if you compile with range checking enabled) occurs on the Count property. Why?


The problem is that GlobalAlloc is happy to allocate more than 64K—that’s why the Size parameter is a LongInt. It’s the pointers that are limited to 64K segments.


In real mode and protected mode, pointers are composed of two components. The least significant component, 16 bits wide, is the offset. It specifies the number of bytes the address is offset from the place indicated by the most significant component, as shown by Figure 9.3.

Figure 9.3: Segmented pointers have two components, offset and segment or selector.


In real mode, the segment is just a paragraph address in which a value of one equals address location 16. In protected mode, this part of the address is called a selector because it selects a 48-bit address descriptor from an array of descriptors. It is not called an index because �it is not a simple, scalar value. The three least significant bits of a selector are used for another purpose entirely and don’t affect which descriptor the selector accesses. Therefore, in a set of possible selectors, the four shown in Figure 9.4 all point to the same segment.


The next selector in the sequence, $1354, would point to a different segment.


Consider the following code fragment, where P is a pChar:





P := Ptr ($1356, $FFFF);


P := @P[1];





If you inspect the machine code emitted for these instructions, you find that no provision has been made for the case where adding 1 to $FFFF rolls a 16-bit value around to zero. It simply happens, causing the word counting method to re-examine the beginning of the text, over and over.


And yet—GlobalAlloc does allocate more than 64K if asked, and it is possible to receive a larger-than-64K block from the Clipboard. So how does this happen—and how can we take advantage of it?


When you ask GlobalAlloc for a “huge” block of memory, Windows first determines how many 64K segments you need. The segments need not be contiguous, but the selectors must be; if you need five 64K segments to fill the request, Windows will have to find five adjacent unused descriptors out of its universe of 8,192 of them. It then causes the adjacent descriptors to point to the necessary segments. (If it can’t find enough adjacent descriptors it will return a handle value of zero.)


When you invoke GlobalLock, you are returned a pointer that includes the first descriptor. You are on your own to figure out how to derive the additional pointers; and you absolutely must not attempt to access any memory chunks that straddle the 64K boundary. Besides the fact that the hardware won’t support it, remember the segments may not be physically adjacent.


To calculate the value of the next selector in the series, you just add eight (23) to the value of the previous selector.


Knowing this, we can write the following function:





Function IncHugePtr (P: Pointer; L: LongInt): Pointer;


	Var


		NewSeg: Word;


		NewOfs: LongInt;


	Begin


	NewSeg := Seg (P^);


	NewOfs := Ofs (P^) + L;


	If NewOfs > $FFFF then


		Begin


		Inc (NewSeg, 8);


		Dec (NewOfs, $10000);


		End;


	IncHugePtr := Ptr (NewSeg, NewOfs);


	End;





Then, invoke it from CountWords instead of using the simplistic pointer incrementation we used before:





Procedure tMainDlg.CountWords (h: tHandle);


	Var


		P: pChar;


			 


			 


	Begin


			 


			 


	While P^ <> #0 do


		Begin


			 


			 


		P := IncHugePtr (P, 1);


		End;


			 


			 


	End;





That’s it: That’s all it takes to implement huge pointers in Borland Pascal. Of course, we needed the cooperation of Windows to get the huge block of memory in the first place. But Word Counter can now safely count words in any block Windows is able to supply.


There may be times when you’d like to work with huge blocks of your own, and there’s no reason why you shouldn’t. For example, in Word Counter we’d like to be able to count words in a file, as well as from the Clipboard; but files are just as likely to run longer than 64K characters as text from the Clipboard. Besides, we’d really like to make the file data look as much like data from the Clipboard as possible so we won’t have to change the CountWords method.


This is not a problem. We’ll just have to implement a FileOpen method, which tDlgAppWindow will invoke if the user chooses the File..Open command, or starts Word Counter with a filename on the command line, or even drags a file on top of the Word Counter window:





Procedure tMainDlg.FileOpen;


	Var


		h: tHandle;


		Size: LongInt;


			 


			 


	Begin


	Inherited FileOpen;


	CountText.SetTextN (0);


	Display.SetText (CountText);


	Size := DataFile.GetSize;


	h := GlobalAlloc (gmem_Moveable, Size+1);


	If h <> 0 then


		Begin


		WaitCursor.Start;


		Help.SetTextR (str_Loading);


		LoadFile (h);


		CountWords (h);


		GlobalFree (h);


		WaitCursor.Complete;


		End


	else


		ErrorBox (str_OutOfMemory);


	End;





The ellipses suggest the presence of LoadFile, a nested procedure. As usual, I’d like to present the context in which LoadFile is called first. �As you can see, there is housekeeping quite similar to that which cmEditPaste carries out: The wait cursor is displayed, and the count display is zeroed. We also obtain the size of the file we intend to load and invoke GlobalAlloc to give us a buffer large enough to store it all.


GlobalAlloc might fail. If you try to load a 13Mb file and you’ve got 3Mb of memory, it will fail, returning a handle value of zero. If you pass that to GlobalLock, you’ll get back a Nil pointer. Try to use that and your application will crash. Checking the handle for validity is smarter.


Once we have that buffer, we can invoke LoadFile. When that procedure returns control, the handle reference to the memory block will look just like the Clipboard had created it; we can pass it on to CountWords. The only difference is that when CountWords is through with the handle, we must free the memory block, because the Clipboard doesn’t own it—we do.


Now that you have the big picture, let’s inspect the nested LoadFile procedure:





Procedure LoadFile (h: tHandle);


	Var


		FCB: File;


		Result: Word;


		P: pChar;


	Begin


	Assign (FCB, DataFile.Pathname.PString);


	Reset (FCB, 1);


	P := GlobalLock (h);


	While not Eof(FCB) do


		Begin


		BlockRead (FCB, P^, $FFFF, Result);


		P := IncHugePtr (P, Result);


		If not Eof(FCB) then


			Begin


			BlockRead (FCB, P^, 1, Result);


			P := IncHugePtr (P, Result);


			End;


		End;


	P^ := #0;


	GlobalUnlock (h);


	Close (FCB);


	End;





It is not possible to read a 64K-sized block; the maximum read count is one less than that. So we read as much as we can ($FFFF is the easiest way to specify the amount) and then, if we did not encounter the end-of-file, read the additional byte. Remember, we cannot straddle a 64K boundary. You could also do two 32K reads; I can’t see much difference. Either way, the transfer of data takes place quickly. When we’ve read all there is to read, we tag a quick NULL at the end of the data, unlock the handle, and close the file.


We said when Word Counter could read files it would be complete. It’s funny how our expectations increase, isn’t it? Because now, wouldn’t it be really neat if Word Counter could be controlled from a word processor, which could send it the text, let it count the words, then pull back the word count and place it, say, on a manuscript—all �automatically?


Well, we’re working with ObjectWindows, aren’t we? In the next chapter, we’ll cover Dynamic Data Exchange and implement those very features.


