If you first programmed PCs that used DOS, you may have become discouraged at the lack of support for serial communications. In fact, a cottage industry arose that specialized in serial communications libraries—and even they didn’t make the programmer’s job easy. There were interrupts and hardware vagaries, and the task was usually so daunting that most programmers just didn’t bother.


Windows, and particularly Windows 3.1, makes accessing the PC’s serial communications ports much simpler. In this chapter, we’ll write an applet that sets your PC’s clock directly from the National Institute of Standards and Technology’s (NIST) computer. In doing so, we’ll also work with transfer buffers and data validation, and use the Borland Workshop Custom Controls.


Making Communication Ports Work for You�tc "Making Communication Ports Work for You"�


In this section, we’ll learn how Windows can be used to access serial ports. Then, we’ll create a tModem class that encapsulates the required code, making it easy to add serial communications to your next project.


Understanding Serial Communications�tc "Understanding Serial Communications"�


The serial communications chip is at the heart of all asynchronous communications. Knowing what it does will help you make better use of it.


Almost all PCs are outfitted with serial ports; if the supporting chips are not on the motherboard, they can be found in an expansion card. Internal modems include them; external modems must be plugged into them.


The chip that controls a serial port does not require the CPU to wait while it does its work. Most of the time the chip simply waits for a character to arrive or be sent. When the wait is over, the chip signals the CPU via an interrupt line that it is ready for its next assignment.


The typical serial chip contains several registers, or hardware variables, that control its operation. There are a pair of small buffers; typically the chip can be fed as many as sixteen bytes at once. The chip then transmits the bytes one at a time, with each byte transmitted one bit at a time. The chip can also receive information; this operation is the inverse of transmission, where bits received one at a time are stored as bytes in the receive buffer. As bytes are transmitted or received, the chip signals the CPU. It is up to the CPU (and the program that is running) to respond appropriately.


The Windows serial communications driver, COMM.DRV, adds a layer of convenience to this by allowing you to initialize, access, and close the port via high-level function calls. COMM.DRV also maintains a pair of buffers, one for input and one for output, which you supply. Since the hardware buffers built into the serial communications chip are only a few bytes wide, working with buffers up to 64K in size is quite convenient—but they are still buffers, and they are still filled asynchronously. You have to check every so often to see if the data you’ve been waiting for has arrived, or if the last byte of the data you’re sending has gone.


Prior to Windows 3.1, you were on your own to figure out a way to keep checking. Most applications used the Windows timer to send themselves messages every few seconds. After receiving the message, they would check the buffers and respond accordingly. But since there was a limited number of timers available, the programmers then had to add code that would verify the availability of a timer, and deal with the fact that one might not be available. Windows 3.1 eased this burden by adding a new message, WM_COMMNOTIFY, which the communications system can generate whenever an event occurs.


Opening the serial port is a little more complex than opening a file. For example, the desired baud rate must be specified, as well as the number of stop bits and the type of parity. The serial chip needs this information to do its job. There are a number of other options, some of them truly arcane. It’s best to assign default values to these unless you really know what you’re doing.


Writing the tModem Class�tc "Writing the tModem Class"�


Modems are used to connect pairs of computers, usually over standard communications channels, such as telephone lines. Although there are synchronous modems, by far the most common are asynchronous models, which either plug into a standard RS-232 port or include a serial communications port as part of their hardware. The tModem class encapsulates the requirements of the Windows serial driver, making it easier to add communication port access to an ObjectWindows �application.


To create the Modem unit, copy CLASSKEL.PAS as MODEM.PAS. The tModem class represents an ephemeral object that cannot be saved or streamed, so you can remove the Load and Store methods and the registration code. The Interface section should be made to look like this:





Unit Modem;





	(***********************************************)


						Interface


	(***********************************************)





	Uses


		WinTypes,


		WinProcs,


		Objects,


		OWindows,


		OString;





	Type


		tConnectTypes =


			(


			Connect,


			Connect_300,


			Connect_1200,


			Connect_2400,


			Connect_4800,


			Connect_9600,


			Connect_Busy,


			Connect_NoDialTone,


			Connect_Error,


			Connect_NoAnswer,


			Connect_NoCarrier,


			Connect_Ring,


			Connect_Timeout,


			Connect_UnknownError


			);





	Type


		pModem = ^tModem;


		tModem = Object (tObject)


			CommID: Integer;


			Baudrate: Word;


			ByteSize, StopBits, Parity: Byte;


			Constructor Init


				(


				aPort: Byte;


				InBufSize, OutBufSize: Word;


				aBaudRate: Word;


				aByteSize, aStopBits, aParity: Byte


				);


			Procedure GetParameters (var DCB: tDCB); Virtual;


			Destructor Done; Virtual;


			Procedure Error (Code: Integer); Virtual;


			Procedure SetText_ (Text: pChar; Length: Word);


			Procedure SetText (const Text: tOString);


			Procedure SetTextP (const Text: String);


			Procedure Yield;


			Procedure Delay (WaitTime: LongInt);


			Function GetText


				(


				var Text: tOString;


				MaxRead: Word;


				MaxTime: LongInt


				): Boolean;


			Function GetLine


				(


				var Text: tOString;


				MaxTime: LongInt


				): Boolean;


			Procedure Reset;


			Procedure Dial (var Number: tOString; Tone: Boolean);


			Function WaitForConnect: tConnectTypes;


			Procedure HangUp;


			End;





The tModem class is designed to be useful in and of itself; but you may want to derive a class from it to use the more esoteric modem parameters. You can set these to non-standard values by overriding GetParameters with a method that calls its ancestor, then makes whatever other settings are required.


tModem.Init actually opens the port:





Constructor tModem.Init


		(


		aPort: Byte;


		InBufSize, OutBufSize: Word;


		aBaudRate: Word;


		aByteSize, aStopBits, aParity: Byte


		);


	Var


		CommName: tOString;


		DCB: tDCB;


	Begin


	Inherited Init;


	CommName.Init (4);


	CommName.SetTextP ('COM');


	CommName.AppendP (Char (Byte ('0') + aPort));


	CommID := OpenComm (CommName.CString, InBufSize, OutBufSize);


	If CommID < 0 then


		Begin


		Error (CommID);


		CommID := 0;


		End;


	CommName.Done;


	BaudRate := aBaudRate;


	ByteSize := aByteSize;


	StopBits := aStopBits;


	Parity := aParity;


	GetParameters (DCB);


	SetCommState (DCB);


	Reset;


	End;





OpenComm is the Windows API call that actually accesses the port. SetCommState uses a record of tDCB structure to set the myriad of port settings:





Procedure tModem.GetParameters (var DCB: tDCB);


	Begin


	DCB.ID := CommID;


	DCB.Baudrate := BaudRate;


	DCB.ByteSize := ByteSize;


	DCB.Parity := Parity;


	DCB.StopBits := StopBits;


	DCB.RlsTimeout := 0;


	DCB.CtsTimeout := 0;


	DCB.DsrTimeout := 0;


	DCB.Flags := $1801;


	DCB.XonChar := #0;


	DCB.XoffChar := #0;


	DCB.XonLim := 10;


	DCB.XoffLim := 10;


	DCB.PeChar := #0;


	DCB.EofChar := #0;


	DCB.EvtChar := '>';


	DCB.TxDelay := 0;


	End;





The first five fields are supplied by the tModem object; the rest are given standard values.


The destructor attempts to leave the modem in its default state by invoking the tModem.Reset method; it then closes the port:





Destructor tModem.Done;


	Begin


	CloseComm (CommID);


	Inherited Done;


	End;





The Error method does nothing; you should override it by a derived class if you want, say, to display a message box should an error occur:





Procedure tModem.Error (Code: Integer);


	Begin


	End;





To make the interface to each class as consistent as possible, we’ve given tModem a set of SetText methods. The base method is SetText_:





Procedure tModem.SetText_ (Text: pChar; Length: Word);


	Var


		State: tComStat;


	Begin


	WriteComm (CommID, Text, Length);


	Repeat


		Begin


		GetCommError (CommID, State);


		Yield;


		End


	Until State.cbOutQue = 0;


	End;





Although serial communication is, by its nature, asynchronous, from the standpoint of a single application it’s usually more convenient to wait for an operation to complete before continuing. Thus, after using the API call WriteComm to send a set of characters to the port, GetCommError is called repeatedly until it reveals that all the characters have been sent. Yield is a tModem method similar to the tDlgAppWindow method of the same name. The parent window is disabled for the duration of the operation, so the user can’t accidentally select a menu item while this method is still executing.


I’ve supplied two SetText methods; you may wish to implement the full set:





Procedure tModem.SetText (const Text: tOString);


	Begin


	SetText_ (Text.CString, Text.Length);


	End;





Procedure tModem.SetTextP (const Text: String);


	Begin


	SetText_ (@Text[1], Word (Text[0]));


	End;





The Yield method simply allows other processing to continue while this application waits for serial port I/O to complete:





Procedure tModem.Yield;


	Var


		Msg: tMsg;


	Begin


	While PeekMessage (Msg, 0, 0, 0, pm_Remove) do


		If not Application^.ProcessAppMsg (Msg) then


			DispatchMessage (Msg);


	End;





Likewise, serial communications frequently need to delay for a measured period of time:





Procedure tModem.Delay (WaitTime: LongInt);


	Var


		Time: LongInt;


	Begin


	Time := GetTickCount;


	Repeat


		Yield


	Until (GetTickCount > (Time + WaitTime));


	End;





GetTickCount is a Windows API call that returns the number of clock ticks since this session of Windows began. You would have to hold a continuous session for about a month and a half before the number rolled over to zero, so it makes an excellent choice for simple timing feats like this. Yield is invoked so other applications don’t come to a halt while this application delays.


There are a dozen transfer protocols commonly used to transmit binary files from one computer to another. A description of each could fill additional chapters, but any would use the GetText method as a base:





Function tModem.GetText


		(


		var Text: tOString;


		MaxRead: Word;


		MaxTime: LongInt


		): Boolean;


	Var


		State: tComStat;


		Time: LongInt;


	Begin


	Time := GetTickCount;


	Text.SetMaxLength (MaxRead);


	Repeat


		Begin


		GetCommError (CommID, State);


		Yield;


		End


	Until (State.cbInQue >= MaxRead) or


		(GetTickCount > (Time + MaxTime));


	Text.Length := ReadComm (CommID, Text.CString, MaxRead);


	Text.CString[Text.Length] := #0;


	End;





In this method, GetCommError is invoked until the required number of characters have arrived, or the maximum wait has been exceeded.


For smaller amounts of text, ASCII transfers are commonly used, complete with carriage-return/line feed terminations. The GetLine method deals best with these:





Function tModem.GetLine


		(


		var Text: tOString;


		MaxTime: LongInt


		): Boolean;


	Var


		State: tComStat;


		Time: LongInt;


		i: Word;


		c: Char;


	Begin


	Time := GetTickCount;


	Text.Clear;


	Repeat


		Begin


		GetCommError (CommID, State);


		For i := 1 to State.cbInQue do


			Begin


			ReadComm (CommID, @c, 1);


			If c in [ #10, #13 ] then


				Begin


				GetLine := True;


				Yield;


				ReadComm (CommID, @c, 1);


				If not (c in [ #10, #13 ]) then


					UngetCommChar (CommID, c);


				Exit;


				End;


			Text.AppendChar (c);


			End;


		Yield;


		End


	Until GetTickCount > (Time + MaxTime);


	GetLine := False;


	End;





In this method, characters are obtained one at a time until the carriage return or line feed is found. A check is made for either because either may be used. Often, but not always, the complementary character will be transmitted next, so the method checks to see if this has happened. The API call UngetCommChar can put the retrieved character back into the buffer if it turns out to be neither a carriage return nor a line feed.


Using Command Mode and Online Mode�tc "Using Command Mode and Online Mode"�


The Hayes Corporation made the PC modem popular; the strong�est feature of the early Hayes modems was their command set. The ability to dial, answer, and connect via standard phone lines was invoked by command, and other modem manufacturers were quick to copy the syntax. The vast majority of PC modems are now Hayes-compatible, meaning they support the basic Hayes command set. (Some modems feature additional commands to support enhanced or unique features.) 


A Hayes-compatible modem can be in one of two states: Command Mode or Online Mode. When your application wishes to use the modem, it has no way of knowing what state the modem is in. 


Windows will prevent two applications from opening a modem at the same time, but it can’t force the previous opener to leave the modem in Command Mode. Thus the first thing your application has to do (after opening the port) is to reset the modem. 





Procedure tModem.Reset;


	Var


		Buffer: tOString;


		Tries: Byte;


	Begin


	FlushComm (CommID, 0);


	FlushComm (CommID, 1);


	Buffer.Init (32);


	Tries := 1;


	Repeat


		Begin


		If Tries = 1 then


			Begin


			Delay (1000);


			SetTextP ('+++');


			Delay (500);


			End;


		FlushComm (CommID, 1);


		SetTextP ('ATZE0V1'+CRLF);


		GetLine (Buffer, 5000);


		GetLine (Buffer, 5000);


		GetLine (Buffer, 5000);


		GetLine (Buffer, 5000);


		Inc (Tries);


		End


	Until Buffer.MatchesP ('OK') or (Tries > 3);


	Buffer.Done;


	End;





If the modem is in Online Mode, sending it the three plus signs, with a one-second delay before and a half-second delay after, will switch it to Command Mode. Almost all the commands in the Hayes command set must be prefixed with “AT” (for ATtention), although you can put several commands on one line as we’ve done here. The “Z” command tells the modem to reset. “E0” tells it not to echo locally typed characters (such as this command). “V1” tells it to respond to commands in words.


The Dial method instructs the modem to dial a phone number:





Procedure tModem.Dial (var Number: tOString; Tone: Boolean);


	Var


		Response: tOString;


	Begin


	Response.Init (32);


	If Tone then SetTextP ('ATDT') else SetTextP ('ATDP');


	SetText (Number);


	SetTextP (CRLF);


	GetLine (Response, 5000);


	Response.Done;


	End;





The Tone parameter lets you specify either a Touch-Tone or rotary phone line.


The WaitForConnect method would likely be called after Dial:





Function tModem.WaitForConnect: tConnectTypes;


	Var


		Response: tOString;


	Begin


	Response.Init (32);


	Response.CaseSensitive := False;


	Repeat


		If not GetLine (Response, 60000) then


			Begin


			WaitForConnect := Connect_Timeout;


			Exit;


			End


	until Response.Length > 0;


	If Response.MatchesP ('Connect') then


		WaitForConnect := Connect


	else if Response.MatchesP ('Connect 300') then


		WaitForConnect := Connect_300


	else if Response.MatchesP ('Connect 1200') then


		WaitForConnect := Connect_1200


	else if Response.MatchesP ('Connect 2400') then


		WaitForConnect := Connect_2400


	else if Response.MatchesP ('Connect 4800') then


		WaitForConnect := Connect_4800


	else if Response.MatchesP ('Connect 9600') then


		WaitForConnect := Connect_9600


	else if Response.MatchesP ('Busy') then


		WaitForConnect := Connect_Busy


	else if Response.MatchesP ('No Dialtone') then


		WaitForConnect := Connect_NoDialtone


	else if Response.MatchesP ('Error') then


		WaitForConnect := Connect_Error


	else if Response.MatchesP ('No Answer') then


		WaitForConnect := Connect_NoAnswer


	else if Response.MatchesP ('No Carrier') then


		WaitForConnect := Connect_NoCarrier


	else if Response.MatchesP ('Ring') then


		WaitForConnect := Connect_Ring


	else


		WaitForConnect := Connect_UnknownError;


	Response.Done;


	End;





A typical modem-using application would send or receive some data, then “hang up” the phone line:





Procedure tModem.HangUp;


	Var


		Response: tOString;


	Begin


	Response.Init (32);


	Delay (1000);


	SetTextP ('+++');


	Delay (500);


	FlushComm (CommID, 0);


	FlushComm (CommID, 1);


	SetTextP ('ATH0' + CRLF);


	GetLine (Response, 1000);


	Response.Done;


	End;





This method will generally be invoked when the modem is in Online Mode; sending the three plus signs returns it to Command Mode. The “H0” command causes the modem to go “on-hook.” 


Writing the SetTime Applet�tc "Writing the SetTime Applet"�


The SetTime applet calls the NIST computer, downloads the correct Greenwich Mean Time (GMT) from it, adjusts for the local time zone, and uses the result to set the system clock. Such an application would not prove much of a challenge to write, except for connecting to the NIST’s computer. But, with the tModem class, even that part is easy!


Creating the Resource Pool�tc "Creating the Resource Pool"�


SetTime’s resource pool must contain the usual assortment of icons, dialog templates, and menus. Here are the required contents.


The SetTime applet will be the first we’ve written to use the Borland Workshop Custom Controls. The application includes one of the Borland buttons; we’ll have to supply a set of bitmaps for it, one each for Up, Down, and Focused. The bitmaps must be numbered, not named, and the number bears a relationship to the button’s control ID, as follows:





•	ID+1000: VGA/Up





•	ID+2000: EGA/Up





•	ID+3000: VGA/Down





•	ID+4000: EGA/Down





•	ID+5000: VGA/Focused





•	ID+6000: EGA/Focused





Generally, the EGA versions are designed to accommodate the EGA’s pixels, which are taller than they are wide, and its more limited palette of eight colors. If you are confident your application will not run on an EGA monitor, or you don’t mind the buttons not having bitmaps if it is used on one, you can omit those bitmaps. The bitmaps I designed, shown in Figure 12.1, are numbered 1010, 3010, and 5010, respectively, based on the button’s ID of 10.

Figure 12.1 The Up, Down, and Focused bitmaps for SetTime's Borland Button.


Since SetTime is one of those rarities, an application that does not open a file, its menu is very simple:





MAIN MENU PRELOAD MOVEABLE DISCARDABLE 


BEGIN


	POPUP "&Setup"


	BEGIN


		POPUP "&Port"


		BEGIN


			MENUITEM "Com&1:", 501


			MENUITEM "Com&2:", 502


			MENUITEM "Com&3:", 503


			MENUITEM "Com&4:", 504


		END


		MENUITEM SEPARATOR


		MENUITEM "E&xit", 24340


	END


	POPUP "&Help"


	BEGIN


		MENUITEM "&Procedures", 904


		MENUITEM SEPARATOR


		MENUITEM "&About...", 999


	END


END





The Stringtable contains the usual online help tag lines for each control:





STRINGTABLE LOADONCALL MOVEABLE DISCARDABLE 


BEGIN


	10, "Press to call the Institute and set your computer's clock."


	1001, "Use format 1-###-###-#### or ###-####."


	1002, "Select your time zone."


	1003, "Check if your community uses Daylight Savings Time."END





In addition, there are a series of strings that describe the various states of the applet as it operates:





STRINGTABLE LOADONCALL MOVEABLE DISCARDABLE 


BEGIN


	12001, "Initializing modem..."


	12002, "Dialing number..."


	12003, "Line is busy. Will retry in one minute."


	12004, "Line is busy. Try again later."


	12005, "Call complete."


	12006, "Operation was unsuccessful."


	12007, "Waiting for connection..."


END





The few Accelerators support the Help system:





MAIN ACCELERATORS PRELOAD MOVEABLE 


BEGIN


	VK_F1, 901, VIRTKEY


	VK_F1, 998, VIRTKEY, CONTROL


	VK_F1, 997, VIRTKEY, SHIFT


END





We should also include our Help Mode cursor.


The icon for SetTime is pictured in Figure 12.2.

Figure 12.2 The icon for SetTime.


Figure 12.3 shows the main dialog. If you’re wondering how to get that “Borland look” in the Resource Workshop, don’t bother. You can use Borland controls, but you’ll have to wait until you actually run the application to see the dialog box background painted gray. 

Figure 12.3 The SetTime dialog template.


Writing the SetTime Framework�tc "Writing the SetTime Framework"�


SetTime is a basic tDlgAppWindow derivative. To make it interesting, it includes the Borland Custom Controls and a tModem object.


Create SETTIME.PAS by copying DLGSKEL.PAS. It should be made to look like this:





Program SetTime;





	{$R SetTime.res}





	Uses


		Objects,


		OWindows,


		WinTypes,


		WinProcs,


		WinDOS,


		Strings,


		DlgApp,


		Controls,


		OString,


		Modem,


		IniData,


		BWCC,


		Validate,


		Menus;





	Const


		cm_Port1 = 101;


		cm_Port2 = 102;


		cm_Port3 = 103;


		cm_Port4 = 104;





	Const


		id_Number = 1001;


		id_TimeZone = 1002;


		id_DaylightSavings = 1003;


		id_Dial = 10;





	Const


		str_Initializing = 12001;


		str_Dialing = 12002;


		str_BusyRetrying = 12003;


		str_BusyNoMoreRetries = 12004;


		str_Complete = 12005;


		str_Failed = 12006;


		str_WaitingForConnect = 12007;





	Type


		pDialCommand = ^tDialCommand;


		tDialCommand = Object (tXButton)


			Procedure bnClicked (var Msg: tMessage);


				Virtual nf_First + bn_Clicked;


			End;





	Type


		pMainDlg = ^tMainDlg;


		tMainDlg = Object (tDlgAppWindow)


			Ini: tIniData;


			Number: tXEdit;


			Timezone: tXCombobox;


			DaylightSavings: tXCheckbox;


			Dial: tDialCommand;


			Port: Byte;


			Constructor Init;


			Destructor Done; Virtual;


			End;





	Type


		pDlgApp = ^tDlgApp;


		tDlgApp = Object (tApplication)


			Procedure InitInstance; Virtual;


			Procedure InitMainWindow; Virtual;


			Function ProcessAppMsg (var Message: tMsg): Boolean; virtual;


			End;





	Constructor tMainDlg.Init;


		Begin


		Inherited Init;


		Ini.Init;


		Number.InitResource (@Self, id_Number, 24);


		Number.Help := @Help;


		Timezone.InitResource (@Self, id_Timezone, 24);


		Timezone.Help := @Help;


		DaylightSavings.InitResource (@Self, id_DaylightSavings);


		DaylightSavings.Help := @Help;


		Dial.InitResource (@Self, id_Dial);


		Dial.Help := @Help;


		End;





	Destructor tMainDlg.Done;


		Begin


		Ini.Done;


		Number.Done;


		Timezone.Done;


		DaylightSavings.Done;


		Dial.Done;


		Inherited Done;


		End;





	Var


		MyDlgApp: tDlgApp;





	Procedure tDlgApp.InitInstance;


		Begin


		Inherited InitInstance;


		hAccTable := LoadAccelerators (hInstance, 'MAIN');


		End;





	Procedure tDlgApp.InitMainWindow;


		Begin


		MainWindow := New (pMainDlg, Init);


		End;





	Function tDlgApp.ProcessAppMsg (var Message: tMsg): Boolean;


		Begin


		ProcessAppMsg :=


			ProcessAccels (Message) or


			ProcessDlgMsg (Message);


		End;





	Begin


	MyDlgApp.Init ('SetTime');


	MyDlgApp.Run;


	MyDlgApp.Done;


	End.





This is the standard setup; it should look quite familiar to you by now. The application creates a main window of tMainDlg type; that type has several controls represented as properties of tMainDlg. These properties are initialized by Init and destroyed by Done. You can compile and run the application; it won’t do anything useful, but it does look pretty.


One reason it looks so pretty is the inclusion of BWCC in the Uses clause. This is the Borland Workshop Custom Control unit. By simply including it like this, the main dialog will have that Borland “chiseled steel” look that many programmers prefer.


Using the Transfer Mechanism�tc "Using the Transfer Mechanism"�


In previous applications of this type, we’ve used a SetupWindow method to load the initial information into the dialog controls. This time, we’re going to use the transfer mechanism, a more convenient and powerful way to move information between controls and memory.


SetTime’s controls aren’t particularly interrelated. If the phone number changes, that won’t affect the time zone or whether the community uses Daylight Savings Time. So why did we bother to instantiate control objects to represent each of them?


The answer is that the transfer mechanism uses these objects to control where and in what order data is to be transferred to the controls. For example, the Number edit control was given a length of 24. This number is only used by the transfer mechanism to specify how many characters to transfer.


In short, the transfer mechanism moves data from a memory block into the dialog controls when the window is displayed, and from the controls back into memory when the box’s OK button is pressed. The transfer can also take place whenever you like, under program control.


The memory location for the data is any block that exists for the life of the dialog box. Each control type requires a specific format of transfer data. tStatic and tEdit objects, for instance, require an array of characters equal in length to the number of characters specified when the objects were constructed. A tListbox requires a pStrCollection object. If it is a single-selection list box, the pStrCollection must be followed by an integer containing the index of the item to be initially selected, or a -1 if none of the items is to be selected. A multi-selection list box requires a selection list in the form of a pointer to a tMultiSelRec.


Derived control objects can replace their inherited Transfer method, thus changing the rules. Since we’re using control objects either from or based on classes from the Controls unit, we can make maintenance of the transfer buffer easier by using tOStrings for tXStatic and tXEdit controls. Open CONTROLS.PAS (or a copy for this chapter) and add the following methods (both Virtual):





Function tXStatic.Transfer


		(DataPtr: Pointer; TransferFlag: Word): Word;


	Type


		pTranRec = ^tTranRec;


		tTranRec = Record


			Item: tOString;


			End;


	Var


		Item: pOString;


	Begin


	Item := @pTranRec (DataPtr)^.Item;


	If TransferFlag = tf_GetData then


		GetText (Item^)


	else if TransferFlag = tf_SetData then


		SetText (Item^);


	Transfer := SizeOf (tTranRec);





	End;





Function tXEdit.Transfer


		(DataPtr: Pointer; TransferFlag: Word): Word;


	Type


		pTranRec = ^tTranRec;


		tTranRec = Record


			Item: tOString;


			End;


	Var


		Item: pOString;


	Begin


	Item := @pTranRec (DataPtr)^.Item;


	If TransferFlag = tf_GetData then


		GetText (Item^)


	else if TransferFlag = tf_SetData then


		SetText (Item^);


	Transfer := SizeOf (tTranRec);





	End;





Likewise, the standard list box and combo box transfer involves a pStrCollection object. Since such objects are collections of sorted strings, that means you can never have an unsorted list box or combo box using the standard transfer mechanism. However, we’re not using standard tCombobox objects; we’re using tXCombobox so we can easily override Transfer with a replacement that will use either a pCollection or pSortedCollection—you choose—of pOStrings:





Function tXCombobox.Transfer


		(DataPtr: Pointer; TransferFlag: Word): Word;


	Type


		pTranRec = ^tTranRec;


		tTranRec = record


			Strings: pCollection;


			Selection: tOString;


			End;


	Procedure DoAdd (P: pOString); far;


		Begin


		AddString (P^.CString);


		End;


	Begin


	If TransferFlag = tf_GetData then


		pTranRec (DataPtr)^.Selection.SetTextW (@Self)


	else if TransferFlag = tf_SetData then


		Begin


		ClearList;


		pTranRec (DataPtr)^.Strings^.ForEach (@DoAdd);


		SetSelString (pTranRec (DataPtr)^.Selection.CString, -1);


		SetText (pTranRec (DataPtr)^.Selection);


		End;


	Transfer := SizeOf (tTranRec);


	End;





The corresponding enhancement to tXListbox has an extra twist: list boxes may allow either single or multiple selections. The tListbox transfer mechanism requires data in different formats depending on the type of list box. I’d prefer a single format—for example, two pCollections: the first containing an object for each list box item, the second in which the items are repeated for any elements that are to be selected. When data is transfered back from the list box, selected items from the first collection will be copied into the second. With that design tXListbox.Transfer looks like this:





Function tXListbox.Transfer


		(DataPtr: Pointer; TransferFlag: Word): Word;


	Type


		pTranRec = ^tTranRec;


		tTranRec = record


			Strings: pCollection;


			Selections: pCollection;


			End;


	Procedure DoAdd (Item: pOString); Far;


		Begin


		AddString (Item^.CString);


		End;


	Procedure DoSelect (Item: pOString); Far;


		Begin


		SetSelIndex (pTranRec (DataPtr)^.Strings^.IndexOf (Item));


		End;


	Var


		Index: Integer;


	Begin


	If TransferFlag = tf_GetData then


		Begin


		pTranRec (DataPtr)^.Selections^.DeleteAll;


		For Index := 0 to pTranRec (DataPtr)^.Strings^.Count - 1 do


			If IsSelected (Index) then


				pTranRec (DataPtr)^.Selections^.Insert


					(pTranRec (DataPtr)^.Strings^.At (Index))


		End


	else if TransferFlag = tf_SetData then


		Begin


		ClearList;


		pTranRec (DataPtr)^.Strings^.ForEach (@DoAdd);


		pTranRec (DataPtr)^.Selections^.ForEach (@DoSelect);


		End;


	Transfer := SizeOf (tTranRec);


	End;





With those enhancements to the Controls unit, we can return to SetTime and add a transfer buffer to tMainDlg. I usually make the transfer buffer part of the tMainDlg object:





Type


	pMainDlg = ^tMainDlg;


	tMainDlg = Object (tDlgAppWindow)


			 


			 


		XBuffer: Record


			Number: tOString;


			TimeZones: pCollection;


			TimeZone: tOString;


			DaylightSavings: Word;


			End;


			 


			 


		Procedure SetInitialData;


		End;





It helps to name the fields in a way that associates them with the control they will feed, but of course that isn’t necessary. What is necessary is that they appear in the same order in which the controls are initialized (not necessarily the same order in which the controls are defined).


Init should invoke SetInitialData and set the address of the transfer buffer:





Constructor tMainDlg.Init;


	Begin


			 


			 


	SetInitialData;


	TransferBuffer := @XBuffer;


	End;





TransferBuffer is an inherited property. There is a flag set by the EnableTransfer and DisableTransfer methods that indicates whether data transfer should take place. However, it is preset for us by our tDlgWindow ancestor; all that’s been lacking is this address.


SetInitialData uses the .INI file to provide the transfer buffer with values:





Procedure tMainDlg.GetInitialData;


	Var


		TZ: pOString;


		z: Word;


	Begin


	Ini.SetPathnameP ('SetTime.INI');


	Ini.SetSectionP ('Set Time');


	Ini.Default.SetTextP ('1-303-494-4774');


	Ini.SetKeyP ('Number');


	XBuffer.Number.InitText (Ini.Value);





	XBuffer.TimeZones := New (pCollection, Init (5, 1));


	XBuffer.TimeZone.Init (0);


	Ini.SetSectionP ('Options');


	Ini.Default.SetTextP ('Eastern');


	Ini.SetKeyP ('Selected');


	XBuffer.TimeZone.SetText (Ini.Value);





	Ini.SetSectionP ('TimeZones');


	Ini.RefreshKeys;


	For z := 1 to Ini.Keys.Count do


		Begin


		TZ := Ini.Keys.At (z-1);


		XBuffer.TimeZones^.Insert (New (pOString, InitText (TZ^)));


		End;





	Ini.Default.SetTextP ('1');


	Ini.SetKeyP ('DaylightSavings');


	XBuffer.DaylightSavings := 


		Word (Ini.Value.CString[0]) - Word ('0');





	Ini.Default.SetTextP ('2');


	Ini.SetKeyP ('Port');


	Port := Word (Ini.Value.CString[0]) - Word ('0');


	End;





Each section (delimited from the others by a blank line) loads a particular variable. Although defaults are provided for each of the fields, that’s not a good way to supply a list of time zones. Here’s an example SETTIME.INI file (which should be placed in the Windows directory):





[Set Time]


Number=1-303-494-4774


DaylightSavings=1


Port=2





[Options]


Selected=Eastern





[TimeZones]


Atlantic=-240


Eastern=-300


Central=-360


Mountain=-420


Pacific=-480


Hawaii=-690





The [TimeZones] section gives the number of minutes by which the time must be offset from Greenwich Mean Time. Minutes are used, rather than hours, because some time zones are not offset by an even number of hours. For example, Hawaii’s offset is -11.5 hours.


Since SetInitialData constructs the objects in XBuffer, they must be destroyed in Done:





Destructor tMainDlg.Done;


	Begin


			 


			 


	XBuffer.Number.Done;


	Dispose (XBuffer.TimeZones, Done);


	XBuffer.TimeZone.Done;


	Inherited Done;


	End;





That’s all it takes. Now when you run SetTime, the phone number, time zone, and Daylight Savings Time controls will be populated and initialized.


Adding Data Validation�tc "Adding Data Validation"�


Data validation is about as easy to add as data transfer. Just create and/or define a validation object, and assign it to the control that is to be validated.


Data validation only applies to edit controls and their derivatives. The best kind of data validation is that which prevents the user from entering incorrect information; it obviates the need for error messages and makes the user feel more secure.


In SetTime, the only field you need to validate is the one containing the phone number for the NIST’s time computer. This field must allow the number to be entered as long distance with area code, and local. (By the middle of 1993 all U.S. phone companies are supposed to convert to this dialing pattern; long distance without area code will no longer be used.) If you’re ambitious, you may want to add variations for pBX codes and credit cards; but we rapidly get to the point where anything is possible and data validation isn’t needed, which kind of defeats the purpose of this exercise.


The tPXPictureValidator class can be used directly. You simply initialize it with a picture string, and it will refuse to allow incorrect input. However, it does this trick silently. A user typing in a hurry might not even notice. So I’ve derived a new class, tPhoneValidator:





Type


	pPhoneValidator = ^tPhoneValidator;


	tPhoneValidator = Object (tPxPictureValidator)


		Function IsValidInput


			(var S: String; SuppressFile: Boolean): Boolean; Virtual;


		End;





The new method just adds a beep when it detects invalid input:





Function tPhoneValidator.IsValidInput


		(var S: String; SuppressFile: Boolean): Boolean;


	Var


		Result: Boolean;


	Begin


	Result := Inherited IsValidInput (S, SuppressFile);


	If not Result then


		MessageBeep (0);


	IsValidInput := Result;


	End;





The validator can be created and assigned in a single statement:





Constructor tMainDlg.Init;


	Begin


			 


			 


	Number.InitResource (@Self, id_Number, 24);


	Number.Help := @Help;


	Number.SetValidator (New (pPhoneValidator,


		Init ('{;1;-###;-###;-####},{###;-####}', True)));


			 


			 


	End;





We’ve provided a picture with two choices: phone numbers in the pattern 1-###-###-#### and ###-#### (where “#” represents a digit). The groups are set off by braces and separated by a comma. When supplying two groups like this, be sure there are enough cues so the validator can tell which one is being used.


The validator will be disposed of automatically when Number is destroyed, so we don’t need to modify the Done method.


Unnatural ACTS�tc "Unnatural ACTS"�


Natural time is a function of the Earth’s rotation, which does not happen at a constant speed. The National Institute of Standards and Technology supplies an approximation of this time via ACTS, the Automatic Computer Time Service.


With any modem-controlling software, such as Windows Terminal, you can access ACTS. The number is (303) 494-4774.


As soon as the connection is made, the following lines appear:





? = HELP


National Institute of Standards and Technology


Telephone Time Service





                        D  L D


 MJD  YR MO DA H  M  S  ST S UT1 msADV        <OTM>


47999 93-04-18 21:39:15 50 0 +.1 045.0 UTC(NIST) *


47999 93-04-18 21:39:16 50 0 +.1 045.0 UTC(NIST) *





If you send a question mark, the time display will cease; in its place ACTS will transmit a help sheet and then disconnect. Otherwise the time lines continue to arrive at the rate of one per second. They are timed so that the asterisk arrives at the midpoint of the time that line describes, offset by a few milliseconds to account for average transmission delays. NIST provides additional means of ensuring utmost accuracy, but using them would be a waste on a PC whose clock is seldom as accurate as a cheap wristwatch. It is that inaccuracy, after all, which gives a program like SetTime its value.


The first step in interpreting the ACTS data is to construct a record template for it:





Type


	t2Digits = Array [1..2] of Char;


	pActsTime = ^tActsTime;


	tActsTime = Record


		JulianDate: Array [1..5] of Char;


		Filler1: Char;


		Year: t2Digits;


		Filler2: Char;


		Month: t2Digits;


		Filler3: Char;


		Day: t2Digits;


		Filler4: Char;


		Hour: t2Digits;


		Filler5: Char;


		Minute: t2Digits;


		Filler6: Char;


		Second: t2Digits;


		Filler7: Char;


		DST: Array [1..2] of Char;


		Filler8: Char;


		LeapSecond: Char;


		Filler9: Char;


		EarthDifference: Array [1..3] of Char;


		Filler10: Array [1..17] of Char;


		OnTimeMarker: Char;


		End;





When the DialCommand button is clicked, tMainDlg must transfer any data changes back to the buffer and kick off the SetTimeFromACTS method:





Procedure tDialCommand.bnClicked (var Msg: tMessage);


	Begin


	pMainDlg (Parent)^.TransferData (tf_GetData);


	pMainDlg (Parent)^.SetTimeFromACTS;


	End;





SetTimeFromACTS allocates a tModem object and connects to ACTS:





Procedure tMainDlg.SetTimeFromACTS;


	Var


		Modem: tModem;


		Acts: tActsTime;


		Year, Month, Day, Hour, Minute, Second: Integer;


	Begin


	Disable;


	Modem.Init (Port, 64, 64, 1200, 8, OneStopBit, NoParity);


	If ConnectToACTS (Modem) then


		If ReadTimeFromACTS (Modem, Acts) then


			Begin


			InterpretTimeFromACTS


				(Acts, Year, Month, Day, Hour, Minute, Second);


			SetDate (Year, Month, Day);


			WinDOS.SetTime (Hour, Minute, Second, 0);


			SendMessage (hwnd Broadcast, wm_TimeChange, 0, 0);


			Help.SetTextR (str_Complete);


			End


		else


			Help.SetTextR (str_Failed);


	Modem.HangUp;


	Modem.Done;


	Enable;


	End;





If the system time has changed, all applications are so notified via the wm_TimeChange message.


If ConnectToACTS is successful, it will return True:





Function tMainDlg.ConnectToACTS (var Modem: tModem): Boolean;


	Var


		Tries: Byte;


		Result: tConnectTypes;


	Const


		MaxTries = 3;


	Begin


	Tries := 0;


	Repeat


		Begin


		DialNumber (Modem);


		Help.SetTextR (str_WaitingForConnect);


		Result := Modem.WaitForConnect;


		Inc (Tries);


		If (Result = Connect_Busy) and (Tries < MaxTries) then


			Begin


			Help.SetTextR (str_BusyRetrying);


			Modem.HangUp;


			Modem.Delay (60000);


			End;


		End


	until (Result <> Connect_Busy) or (Tries = MaxTries);


	If Result = Connect_Busy then


		Begin


		Help.SetTextR (str_BusyNoMoreRetries);


		ConnectToACTS := False;


		End


	else if


			not (Result in [ Connect, Connect_300..Connect_9600 ]) then


		Begin


		Help.SetTextR (str_Failed);


		ConnectToACTS := False;


		End


	else


		ConnectToACTS := True;


	End;





DialNumber does just what it says:





Procedure tMainDlg.DialNumber (var Modem: tModem);


	Var


		Phone: tOString;


	Begin


	Help.SetTextR (str_Initializing);


	Modem.Reset;


	Help.SetTextR (str_Dialing);


	Phone.Init (32);


	Phone.SetTextW (@Number);


	Modem.Dial (Phone, True);


	Phone.Done;


	End;





ReadTimeFromACTS makes as many attempts as reason dictates to get a line of the correct length and ending in “#” or “*”—either of which is possible. The constant MaxTries limits these attempts; I’ve set its value to 30:





Function tMainDlg.ReadTimeFromActs


		(var Modem: tModem; var Acts: tActsTime): Boolean;


	Var


		Buffer: tOString;


		Tries: Word;


		ActsBuffer: pActsTime;


	Const


		MaxTries = 30;


	Begin


	Buffer.Init (SizeOf (tActsTime));


	Tries := 0;


	Repeat


		Begin


		Inc (Tries);


		If Tries > MaxTries then


			Begin


			ReadTimeFromActs := False;


			Exit;


			End;


		Modem.GetLine (Buffer, 2000);


		ActsBuffer := pActsTime (Buffer.CString);


		End


	Until (Buffer.Length = SizeOf (tActsTime)) and


		(ActsBuffer^.OnTimeMarker in ['#', '*']);


	Acts := ActsBuffer^;


	ReadTimeFromActs := True;


	End;





InterpretTimeFromACTS is a bit bulky; it has to convert the ASCII data to binary, apply the time zone offset, then correct for changes in day, month, and year that can occur when the line of midnight lies between your community and Greenwich:





Procedure tMainDlg.InterpretTimeFromACTS


		(


		var Acts: tActsTime;


		var Year, Month, Day, Hour, Minute, Second: Integer


		);


	Function Cvt (d: t2Digits): LongInt;


		Begin


		Cvt :=


			((Word (d[1]) and $000F) * 10) + (Word (d[2]) and $000F);


		End;


	Function Offset: LongInt;


		Var


			Value: LongInt;


			Code: Integer;


		Begin


		Ini.Default.SetTextP ('0');


		Ini.SetKeyW (@TimeZone);


		Val (Ini.Value.PString, Value, Code);


		Offset := Value;


		End;


	Procedure NormalizeTime;


		Const


			DaysInMonth: Array [1..12] of Word =


				(31, 28, 31, 30, 31, 30, 31, 31, 30, 31, 30, 31);


		Function IsLeapYear (Year: Word): Boolean;


			Begin


			IsLeapYear := False;


			End;


		Begin


		While Minute < 0 do


			Begin


			Dec (Hour);


			Inc (Minute, 60);


			End;


		While Minute >= 60 do


			Begin


			Inc (Hour);


			Dec (Minute, 60);


			End;


		While Hour < 0 do


			Begin


			Dec (Day);


			Inc (Hour, 24);


			End;


		While Hour >= 24 do


			Begin


			Inc (Day);


			Dec (Hour, 24);


			End;


		If IsLeapYear (Year) then


			DaysInMonth[2] := 29;


		If Day < 0 then


			Begin


			Dec (Month);


			Day := DaysInMonth[Month];


			End;


		If Day > DaysInMonth[Month] then


			Begin


			Inc (Month);


			Day := 1;


			End;


		If Month < 0 then


			Begin


			Dec (Year);


			Month := 12;


			End;


		If Month > 12 then


			Begin


			Inc (Year);


			Month := 1;


			End;


		End;


	Begin


	Year := Cvt (Acts.Year) + 1900;


	Month := Cvt (Acts.Month);


	Day := Cvt (Acts.Day);


	Hour := Cvt (Acts.Hour);


	Minute := Cvt (Acts.Minute) + Offset;


	Second := Cvt (Acts.Second);


	NormalizeTime;


	End;


Adding the Finishing Touches�tc "Adding the Finishing Touches"�


SetTime must be able to choose which communication port to use. It must also be able to save changes made to its various parameters.


We’re using a nested drop-down menu, both to set the communication port and to display the current choice. To manage check marks, we’ll need a tMenuItem for each of the port menus; to set the communication port, we only have to trap the four menu commands associated with the File..Ports nested menu:





Type


	pMainDlg = ^tMainDlg;


	tMainDlg = Object (tDlgAppWindow)


			 


			 


		Port: Byte;


		PortMenu: Array [1..4] of tMenuItem;


			 


			 


		Procedure SetupWindow; Virtual;


		Procedure cmPort1 (var Msg: tMessage);


			Virtual cm_First + cm_Port1;


		Procedure cmPort2 (var Msg: tMessage);


			Virtual cm_First + cm_Port2;


		Procedure cmPort3 (var Msg: tMessage);


			Virtual cm_First + cm_Port3;


		Procedure cmPort4 (var Msg: tMessage);


			Virtual cm_First + cm_Port4;


		End;





tMenuItem objects can’t be constructed until SetupWindow has been invoked, so we’ll add a SetupWindow method:





Procedure tMainDlg.SetupWindow;


	Var


		h: tHandle;


		p: Word;


	Begin


	Inherited SetupWindow;


	h := GetMenu (hWindow);


	h := GetSubMenu (h, 0);


	h := GetSubMenu (h, 0);


	For p := 1 to 4 do


		PortMenu[p].InitResource (h, p-1);


	PortMenu[Port].Check;


	End;





This method also places the initial check mark by the name of the currently selected port. This value was obtained from the .INI file in SetInitialData. We have to dispose of the tMenuItem properties in Done:





Destructor tMainDlg.Done;


	Begin


			 


			 


	PortMenu[1].Done;


	PortMenu[2].Done;


	PortMenu[3].Done;


	PortMenu[4].Done;


	Inherited Done;


	End;





Each of the four menu handlers unchecks the previous selection, changes the Port property, and checks the new selection. It also writes the new value to the .INI file:





Procedure tMainDlg.cmPort1 (var Msg: tMessage);


	Begin


	PortMenu[Port].Uncheck;


	Port := 1;


	Ini.SetKeyP ('Port');


	Ini.SetValueP ('1');


	PortMenu[Port].Check;


	End;





Procedure tMainDlg.cmPort2 (var Msg: tMessage);


	Begin


	PortMenu[Port].Uncheck;


	Port := 2;


	Ini.SetKeyP ('Port');


	Ini.SetValueP ('2');


	PortMenu[Port].Check;


	End;





Procedure tMainDlg.cmPort3 (var Msg: tMessage);


	Begin


	PortMenu[Port].Uncheck;


	Port := 3;


	Ini.SetKeyP ('Port');


	Ini.SetValueP ('3');


	PortMenu[Port].Check;


	End;





Procedure tMainDlg.cmPort4 (var Msg: tMessage);


	Begin


	PortMenu[Port].Uncheck;


	Port := 4;


	Ini.SetKeyP ('Port');


	Ini.SetValueP ('4');


	PortMenu[Port].Check;


	End;





Rather than updating the .INI file every time a change is made to the phone number or time zone list, we’ll just update it when the user hits the Dial button. That means adding a line to tDialCommand.bnClicked:





Procedure tDialCommand.bnClicked (var Msg: tMessage);


	Begin


			 


			 


	pMainDlg (Parent)^.UpdateFields;


	End;





Then, we’ll add the UpdateFields method to tMainDlg:





Procedure tMainDlg.UpdateFields;


	Begin


	Ini.SetSectionP ('Set Time');


	Ini.SetKeyP ('Number');


	Ini.SetValueW (@Number);


	Ini.SetSectionP ('Options');


	Ini.SetKeyP ('Selected');


	Ini.SetValueW (@TimeZone);


	End;





That completes the SetTime application. Here are some things you might like to add:





•	When the line is busy, you could display a countdown on the status bar instead of a static message. 





•	When the PC’s clock is wrong, you could calculate and display the difference. 





You can probably think of other things yourself. It’s your program. Go ahead and play!


